

**Csupa játék
ZX Spectrumra**

A programokat középiskolás diákok készítették

Megjegyzésekkel látta el és ellenőrizte

Donáth János

Szerkesztette:

Votisky Zsuzsa

© Műszaki Könyvkiadó 1985

ETO 519.683

ISBN: 963 10 6771 8

*Készült a Neumann János Számítástechnikai Társulat támogatásával
a Műszaki Könyvkiadó gondozásában*

Csupa játék

ZX Spectrumra

*A μ Mikroszámítógép Magazin
melléklete*

BEVEZETÉS

Az itt közzétett programokat fiatalok készítették: középiskolás diákok. Életkoruk, valamint a magyarországi mikroszámítástechnika helyzete következtében valószínűleg csak nemrég ismerkedtek meg ennek a tudománynak (és hobbynak) a szépségeivel, hasznosságával, szakmai alapismereteivel. Többségük otthonában — minden bizonnyal — már ott van egy saját családi tulajdonban levő vagy egy kölcsönkért számítógép. Az első fázisban általában azok a — sokszor valóban sziporkázóan szellemes — játékprogramok játsszák a főszerepet, amelyek bőségesen kaphatók a nyugati áruházakban, és sokszor az „anti-számítógépesek”-et is képesek „megfertőzni”. Ez a láz általában — és szerencsére — nem tart soká, mert felmerül a saját elképzelések megvalósításának igénye. Az újdonság varázsában természetes, hogy elsősorban a játékprogramok jönnek számításba. Ez azonban már minőségileg más tevékenységet és gondolatmenetet kíván. Szabad utat kap a fantázia, megkezdődik a programozástechnikai ismeretek gyakorlati elsajátítása. Új generáció ismerkedik a legújabb technikával.

A fiataloknak még nem kell tudniuk, hogy a számítógép bár kitűnően szórakoztat, de alapvetően nem erre való, nem ezért találták fel és fejlesztették ki őket. A számítógép és a számítástechnika alkalmazása — megfelelő előkészületek, gépkiválasztás, hatékony használat mellett — szükségszerű gazdasági, tehát közvetett vagy közvetlen hasznot termelő tevékenység. Mire ezek a fiatalok dolgozni kezdenek, a számítástechnikához vezető úton az első lépéseket megtették, a számítástechnika alapfogalmait megszokták, és otthonosan mozognak köztük.

Most tehát játékprogramok listáit tesszük közzé. Az ötletek persze többé-kevésbé ismerősek is lehetnek. Nem vizsgáltuk, nem is vizsgálhattuk a játékötletek és megvalósításuk „eredetiségét” — de mindegyik program némiképp magán viseli már alkotójának egyéniségét is.

A szerzők túljutottak a programtervezés és a kódolás sokszor unalmas, fáradtságos, igen pontos munkát követelő kezdő lépésein, tesztelték, esetleg módosították, javították programjaikat. Rákényszerültek az ötletek és a fegyelmezett munka összeegyeztetésére, és a problémákkal több-kevesebb sikerrel meg is birkóztak. További eredményes programozói munkát kívánunk nekik és mind-

azoknak, akik talán még nem jutottak el az önálló programkészítéshez, de kedvet, ambíciót éreznek hozzá.

A most induló, és várhatóan folytatódó sorozat első kötete — talán kicsit önkényes választással — a ZX Spectrum személyi számítógépre írt programokat tartalmazza. Ezt a választást a gép viszonylagos elterjedtsége, az egyszerű programozhatóság és a sokféle kedvező tulajdonság (hanghatás, színkezelés stb.) indokolta. A későbbiekben remélhetőleg más számítógéptípusokra írt programok közlésére is lehetőség nyílik.

Lássuk tehát a programokat, amelyeket ki-ki begépelhet a saját gépébe! Közzöljük a program alkotójának nevét, a program rövid tartalmi leírását, a teljes listát és alkalmanként saját szubjektív észrevételeinket. Ez utóbbiakat inkább tanácsnak, mintsem kritikának szánjuk.

Azt javasoljuk, hogy ne sorról sorra másolják be a programot, hanem először „olvassák át” és az elágazásokat követve, egy-egy szubrutin beírásával, kipróbálásával haladjanak előre. Így az egyes módosításokat is könnyebb elvégezni. Még egy szakmai megjegyzés: a Spectrum gépek jelenleg 16 K-s vagy 48 K-s memóriával kerülnek forgalomba.

Mi nem tettünk különbséget a két gépi változat között, de a némileg eltérő memóriafelépítés miatt a fixcímek néha különbözők. Ez az egyes programok esetén címváltozást okozhat elsősorban a programozható karakterek, az UDG-k elhelyezésénél. Csupán az első két program készült 16 K-s gépen. A listák végén a grafikus karaktereket ábécésorrendben többnyire feltüntettük.

KOCKA

Készítette: Kovács Zoltán

Leírás: A program négy színt kér a játékostól. Ebből a négy színből elkészít és a képernyőn megjelenít egy 4X4-es négyzetet, amelynek elemei a megadott színekből alkotott négyzetek. Egy villogó nyíllal lehet pozícionálni az x és z billentyűkkel arra a sorra vagy oszlopra, amelyen egyet – ciklikusan – forgatni akarunk. Ilyenkor az utolsó kis négyzet az első helyére lép, a többi pedig értelemszerűen arrébb csúszik. A teljes 4X4-es négyzetet – kicsit a bűvös kockához hasonlóan – addig kell így „forgatni”, amíg minden vízszintes sorban egyforma színű négyzet lesz. A program a forgatások számát számolja, a helyes megoldáskor zenél, majd újra indítható. A sorok (oszlopok) az N billentyű hatására fordulnak.

```
1>GO TO 1000
5 DIM a(4): DIM b(4): DIM c(8): DIM d(8,4)
6 FOR g=1 TO 4: PRINT g;".SZIN ";
7 INPUT f: IF f<0 OR f>7 THEN GO TO 6
8 LET b(g)=INT f: PRINT f: NEXT g
9 CLS
10 LET l=0: LET q=22827
20 FOR f=11 TO 17 STEP 2
25 FOR g=9 TO 15 STEP 2
30 GO SUB 70
40 NEXT g: NEXT f
45 FOR f=1 TO 8: FOR g=1 TO 4
50 READ a: LET d(f,g)=q+a
55 NEXT g: NEXT f
60 DATA 0,2,4,6,64,66,68,70,128,130,132,134,192,194
,196,198,192,128,64,0,194,130,66,2,196,132,68,4,198,1
34,70,6
65 GO TO 150
70 LET a=INT (RND*4)+1
```

```

80 IF a(a)=4 THEN GO TO 70
90 LET a(a)=a(a)+1
100 LET s=b(a): INK s
110 PRINT AT g,f;"AB"
120 PRINT AT g+1,f;"DC"
130 RETURN
150 INK 0
160 FOR f=9 TO 15 STEP 2
165 LET c((f-7)/2)=22528+f*32+8
170 PRINT AT f,8;"E": NEXT f
180 PRINT AT 18,11;"F F F F"
190 FOR f=5 TO 8
200 LET c(f)=23105+(f*2)
210 NEXT f
215 POKE c(1),231: LET c=1
220 IF INKEY$="x" THEN GO TO 300
225 IF INKEY$="z" THEN GO TO 350
230 IF INKEY$="n" THEN GO TO 400
232 IF PEEK q<>PEEK (q+2) OR PEEK (q+2)<>PEEK (q+4)
OR PEEK (q+4)<>PEEK (q+6) THEN GO TO 240
235 IF PEEK (q+64)<>PEEK (q+66) OR PEEK (q+66)<>PEEK
(q+68) OR PEEK (q+68)<>PEEK (q+70) THEN GO TO 240
237 IF PEEK (q+128)<>PEEK (q+130) OR PEEK (q+130)<>P
EEK (q+132) OR PEEK (q+132)<>PEEK (q+134) THEN GO TO
240
239 GO TO 600
240 GO TO 220
300 POKE c(c),56: LET c=c+1
310 IF c=9 THEN LET c=1
320 POKE c(c),231
325 FOR f=0 TO 30: NEXT f
330 GO TO 220
350 POKE c(c),56: LET c=c-1
360 IF c=0 THEN LET c=8
370 GO TO 320
400 LET e1=d(c,1): LET e2=d(c,2)
410 LET e3=d(c,3): LET e4=d(c,4)
420 LET f1=PEEK e1: LET f2=PEEK e2

```



```

430 LET f3=PEEK e3: LET f4=PEEK e4
440 LET r1=e3: LET r2=f2: GO SUB 500
450 LET r1=e4: LET r2=f3: GO SUB 500
460 LET r1=e1: LET r2=f4: GO SUB 500
470 LET r1=e2: LET r2=f1: GO SUB 500
475 LET l=l+1: PRINT AT 5,14;l
480 GO TO 220
500 POKE r1,r2: POKE r1+1,r2
510 POKE r1+32,r2: POKE r1+33,r2
520 RETURN
600 FOR f=0 TO 10 STEP .1
610 BEEP .01,f: NEXT f
620 PRINT AT 5,12;"ÚJ JATEK-I"
625 PAUSE 100
630 PAUSE 0: IF INKEY$="i" THEN RUN
640 SAVE "Kocka"
650 STOP
1000 FOR f=0 TO 31: READ a: NEXT f
1005 FOR f=32600 TO 32647
1010 READ a: POKE f,a: NEXT f
1015 CLS : PRINT "Torold ki az 1.sort,es futtas  ujb
ol!"
1020 DATA 0,127,127,127,127,127,127,127
1030 DATA 0,254,254,254,254,254,254,254
1040 DATA 254,254,254,254,254,254,254,0
1050 DATA 127,127,127,127,127,127,127,0
1060 DATA 24,12,6,255,255,6,12,24
1070 DATA 24,60,127,219,153,24,24,24

```

Megjegyzés: Szellemes és látványos program. Jól kihasználja a számítógép színkezelését. Valamilyen egyszerű, felhasználóközeli módon megoldható lenne, hogy a színek választékát „menüszerűen” a képernyőn a játékosal tudassa a program, bár tulajdonképpen közömbös a játék szempontjából, hogy melyek a kiválasztott színek. Ezenkívül, ha az „Új játék?” kiírásra nem „i” karakterrel válaszolunk, a számítógép furcsán reagál, nyomban kimentti ugyanis a játékot. A képernyőn közölt információk fogalmazása és helyesírása lehetne pontosabb. Sajnos teljesen hiányoznak a REM-ek a programlistából.

LOGICE

Készítette: Kovács Zoltán

Leírás: Ez a program a „Négy a nyerő” néven is ismert játék számítógépes változata. Mindig a játékos játszik a gép ellen. A játékmező hét oszlopból áll, és csak az oszlopok alsó pozíciójába lehet lépni. Ilyenkor az oszlop eggyel feljebb tolódik. Az egyes lépésekhez tehát a megfelelő számot kell beütni. Az a cél, hogy akár vízszintesen, akár függőlegesen, akár átlósan négy egyforma színű figura kerüljön egymás mellé. Ha sikerül, a gép értékel.

```
1>GO TO 1000
10 DIM f(9): DIM p(9): DIM b(9): DIM c(9)
20 DIM a(11,12)
30 LET p1=0: LET p2=0: LET r5=0
40 CLS : PRINT AT 10,10;"KI KEZDJEN?";AT 12,8;"1-JA
TEKOS;2-GEP"
50 IF INKEY$="1" THEN GO TO 100
60 IF INKEY$="2" THEN GO SUB 200: GO TO 120
70 GO TO 50
100 GO SUB 200
110 INK 5: PRINT FLASH 1;AT 10,0;"JATEKOS";: INK 2:
PRINT FLASH 0;AT 10,26;"GEP";: LET r5=r5+1: IF r5=4
3 THEN GO TO 900
115 GO SUB 300
120 INK 2: PRINT FLASH 1;AT 10,26;"GEP";: INK 5: PR
INT FLASH 0;AT 10,0;"JATEKOS";: LET r5=r5+1: IF r5=4
3 THEN GO TO 900
130 GO SUB 500
190 GO TO 110
200 CLS : FOR f=3 TO 9: FOR g=3 TO 8
210 LET b(f)=3+f*2: LET c(g)=2+g*2: PRINT AT c(g),b(
f);"EF";
215 LET a(g,f)=0
```

```

220 NEXT g: NEXT f
240 RETURN
300 LET a$=INKEY$: IF a$>"0" AND a$<"8" THEN GO TO
320
310 GO TO 300
320 LET a=2+VAL (a$)
330 FOR g=8 TO 3 STEP -1: IF a(g,a)=0 THEN GO TO 35
0
340 NEXT g: GO TO 400
350 INK 5: LET a(g,a)=2: PRINT AT c(g)-1,b(a); "AE"
;: PRINT AT c(g),b(a); "CD";
360 LET f1=a: LET f2=g: LET f5=2: LET g6=1: GO SUB 7
00
370 INK 0: RETURN
400 PRINT AT 1,10;"Hibas lepes!"
410 BEEP 1,1: PAUSE 50: PRINT AT 1,10;" "
420 GO TO 300
500 FOR g=3 TO 9: LET f(g)=0
510 FOR r=3 TO 8: IF a(r,g)=0 THEN LET f(g)=r
520 NEXT r: NEXT g
530 LET g=0: FOR r=3 TO 9: IF f(r)=0 THEN LET p(r)=
-100: GO TO 550
540 LET f1=r: LET f2=f(r): LET g6=0: GO SUB 700: LET
p(r)=g3
550 IF p(r)>g THEN LET g=p(r): LET n=r: LET k=f(r)
560 NEXT r
570 LET a(k,n)=1: INK 2: PRINT AT c(k)-1,b(n); "AE"
580 PRINT AT c(k),b(n); "CD";
590 LET f1=n: LET f2=k: LET g6=1: LET f5=1: GO SUB 7
00
600 INK 0: RETURN
700 LET g3=0: LET f3=1: LET f4=1: GO SUB 710: LET f3
=0: GO SUB 710: LET f3=-1: GO SUB 710: LET f4=0: GO S
UB 710: RETURN
710 FOR z=1 TO 2: LET x=0: LET g4=1
715 FOR c=1 TO 3
720 LET f1=f1+f3: LET f2=f2+f4
725 IF a(f2,f1)<>z THEN GO TO 740

```

```

730 LET g4=g4+1: NEXT c: LET c=3
740 LET f1=f1-c*f3: LET f2=f2-c*f4
750 FOR c=1 TO 3: LET f1=f1-f3: LET f2=f2-f4:
755 IF a(f2,f1)<>z THEN GO TO 770
760 LET g4=g4+1: NEXT c: LET c=3
770 LET f1=f1+c*f3: LET f2=f2+c*f4
780 LET g3=g3+(g4*g4*g4)*(4-z): IF g4>=4 THEN GO TO
800
790 NEXT z: RETURN
800 IF g6=0 THEN GO TO 790
810 IF f5=2 THEN GO TO 850
820 IF f5=z THEN LET p1=p1+1: PRINT AT 12,27;p1;: G
0 TO 790
830 GO TO 790
850 IF z=f5 THEN LET p2=p2+1: PRINT AT 12,5;p2;: GO
TO 790
860 GO TO 790
900 CLS : INK 1: PRINT AT 10,15;p1;"--";p2: IF p1>p2
THEN GO TO 930
910 IF p1<p2 THEN GO TO 950
920 PRINT AT 13,13;"Dontetlen!": GO TO 980
930 PRINT AT 13,10;"Gyozott a gep!"
940 GO TO 980
950 PRINT AT 13,8;"Gratulalok! Nyertel!"
980 FOR f=0 TO 10 STEP .1: BEEP .05,f: NEXT f: STOP
1000 FOR f=32600 TO 32647
1010 READ a: POKE f,a: NEXT f
1015 RUN 10
1020 DATA 0,3,15,31,63,60,120,120,0,192,240,248,252,6
0,30,30,120,120,60,63,31,255,255,255,30,30,60,252,248
,255,255,255
1030 DATA 0,0,0,0,0,112,124,127,0,0,0,0,0,14,62,254

```

Megjegyzés: A játékkal kapcsolatban nincs különösebb megjegyzésünk, szóra-
koztató logikai játék. A listából szinte teljesen hiányoznak a REM-ek, és a
programutasítások felépítése zsúfolt, nehezen áttekinthető (bár sokkal kevésbé,
mint az előző „KOCKA” programban).

BOGARAK

Készítették: Galambos Attila, Rusznyák Gábor, Király László

Leírás: A játék célja, hogy a képernyőre rajzolt hálózat egy véletlen helyén megjelenő villogó csillagot „megegye”, ill. elérje a játékos. A pályán eközben négy bogár mozog, amelyek követik a játékos mozgásának irányát, és megkísérik utolérni. Ha a bogarak bekerítési manővere sikerül, vége a játéknak. Ha ellenben nem, a játékos eléri a villogó csillagot, a program jutalmul zenél és 10 pontot ad. Ilyenkor a cél-csillag, a bogarak, valamint a játékos figurájának a helye véletlenszerűen megváltozik. Egy „élet” van: az elért találatok számát végeredményként kezeli a gép.

Az irányítógombok:

M – jobbra

N – balra

A – felfelé

Z – lefelé

```
1 REM *** BOGARAK ***
5 POKE 23658,24
6 GO SUB 9000: GO TO 3000
7 LET P=0
10 BORDER 1: PAPER 2: INK 0: CLS
15 REM * PALYA *
20 FOR I=0 TO 21: PRINT AT I,0; PAPER 1;" ": NEXT I
30 FOR I=0 TO 21: PRINT AT I,30; PAPER 1;" ": NEXT
I
40 PRINT AT 21,1; PAPER 1;"
"
50 PAPER 7: FOR I=0 TO 20 STEP 4: FOR J=1 TO 29: PR
INT AT I,J;" ": NEXT J: NEXT I
60 FOR I=1 TO 29 STEP 4: FOR J=0 TO 20: PRINT AT J,
I;" ": NEXT J: NEXT I
1000 LET I$= "GFHE": DIM B$(4): LET A$= "A"
```

```

1010 DIM X(4): DIM Y(4): DIM C(4): DIM D(4)
1100 FOR N=1 TO 4: LET X(N)=4*INT (RND*8): LET Y(N)=4
*INT (RND*5): PRINT AT Y(N),X(N)+1;I$(4): NEXT N
1110 LET X=4*INT (RND*8): LET Y=4*INT (RND*5)
1120 FOR N=1 TO 4: IF Y=Y(N) AND X=X(N) THEN GO TO 1
110
1125 NEXT N
1130 PRINT FLASH 1;AT Y,X+1;"A": PAUSE 0: PRINT AT
Y,X+1;"A"
1190 FOR N=1 TO 4: LET C(N)=X(N): LET D(N)=Y(N): NEXT
N
1200 LET ZX=4*INT (RND*8): LET ZY=4*INT (RND*5)
1210 FOR N=1 TO 4: IF (X=X(N) AND X/4=INT (X/4)) OR (
Y=Y(N) AND Y/4=INT (Y/4)) THEN LET C(N)=X: LET D(N)=
Y
1220 PRINT AT Y(N),X(N)+1;" ": LET X(N)=X(N)<(C(N)-X(
N)): LET Y(N)=Y(N)+SGN (D(N)-Y(N))
1230 IF X(N)<>C(N) OR Y(N)<>D(N) THEN GO TO 1300
1240 IF INT (RND+.5) THEN GO TO 1280
1250 LET C(N)=C(N)+4*SGN (RND-.5): IF C(N)<0 THEN LE
T C(N)=C(N)+8
1260 IF C(N)>28 THEN LET C(N)=C(N)-8
1270 GO TO 1300
1280 LET D(N)=D(N)+4*SGN (RND-.5): IF D(N)<0 THEN LE
T D(N)=D(N)+8
1290 IF D(N)>20 THEN LET D(N)=D(N)-8
1300 LET B$(N)=I$((C(N)>X(N))+2*(X(N)>C(N))+3*(D(N)>Y
(N))+4*(Y(N)>D(N)))
1310 IF X=X(N) AND Y=Y(N) THEN GO TO 2000
1320 PRINT AT Y(N),X(N)+1;B$(N): NEXT N: PRINT FLASH
1;AT ZY,ZX+1;"*": PRINT AT Y,X+1;" "
1330 IF INKEY$="A" THEN LET Y=Y-(Y>0)*(X/4=INT (X/4)
): LET A$="A"
1340 IF INKEY$="Z" THEN LET Y=Y+(Y<20)*(X/4=INT (X/4
)): LET A$="D"
1350 IF INKEY$="N" THEN LET X=X-(X>0)*(Y/4=INT (Y/4)
): LET A$="B"
1360 IF INKEY$="M" THEN LET X=X+(X<28)*(Y/4=INT (Y/4
)): LET A$="C"

```

```

1370 PRINT AT Y,X+1;A$: IF X=ZX AND Y=ZY THEN GO TO
1800
1380 GO TO 1210
1800 LET P=P+10: GO SUB 1830: PRINT AT 21,0; PAPER 1;
  INK 6; "          ";P;" PONT"
1810 FOR N=1 TO 4: PRINT AT Y(N),X(N)+1;" ": NEXT N:
PRINT AT Y,X+1;" "
1820 GO TO 1100
1830 FOR I=0 TO 20: BEEP .07,I: NEXT I: FOR I=30 TO 1
0 STEP -1: BEEP .05,I: NEXT I: RETURN
2000 PRINT AT Y,X+1;"*": FOR N=20 TO 0 STEP -1: BEEP
.02,N: IF N=15 THEN PRINT AT Y,X+1;"+"
2010 IF N=10 THEN PRINT AT Y,X+1;". "
2020 NEXT N: PRINT AT Y,X+1;" ": PAUSE 100
2030 PAPER 4: INK 7: FOR I=0 TO 20 STEP 3: FOR J=1 TO
29: PRINT AT I,J;" ": BEEP .01,I+J: NEXT J: NEXT I
2040 FOR I=1 TO 20 STEP 3: FOR J=1 TO 29: PRINT AT I,
J;" ": BEEP .01,I+J: NEXT J: NEXT I
2050 FOR I=2 TO 20 STEP 3: FOR J=1 TO 29: PRINT AT I,
J;" ": BEEP .01,I+J: NEXT J: NEXT I
2060 PRINT AT 9,11;"JATEK VEGE"
2070 PRINT AT 12,9;"PONTSZAM : ";P
2080 PRINT FLASH 1;AT 15,1;"NYOMJ EGY GOMBOT A KEZDE
SHEZ!"
2090 PAUSE 500
3000 PAPER 7: INK 2: CLS : PRINT "*****
*****
          **
      BOGARAK          **
*****"
3010 PRINT : PRINT " Ebben a jatekban a feladatod:
el kell jutnod a villogo csil- lagig ugy, hogy ne eg
yenek meg a bogarak."
3020 PRINT : PRINT " A jatekhoz sok sikert kivannak
a jatek keszitoi:"
3030 PRINT : PRINT " Kiraly Laszlo
Galambos Attila          Rusznyak Gabor"
3040 PAUSE 500: GO TO 7
4000 CLEAR : SAVE "BOGARAK" LINE 1: VERIFY "BOGARAK":
RUN

```

```
8999 PAPER 7: INK 0: STOP
9000 FOR I=0 TO 63: READ A: POKE USR "A"+I,A: NEXT I:
  RETURN
9010 DATA 0,24,60,126,126,102,66,0
9020 DATA 0,30,60,120,120,60,30,0
9030 DATA 0,120,60,30,30,60,120,0
9040 DATA 0,66,102,126,126,60,24,0
9050 DATA 165,153,126,60,255,60,126,153
9060 DATA 201,42,190,127,127,190,42,201
9070 DATA 147,84,125,254,254,125,84,147
9080 DATA 153,126,60,255,60,126,153,165      LLIST
```

Megjegyzés: A játék jó, de kissé nehezen áttekinthető a programlista (pl. logikailag fontos utasítások nem a programsorszám után, első utasításként íródtak, hanem ötletszerűen, szinte folyamatosan). Kicsit feleslegesen hosszú ideig dudorászik és törli a képernyőt a program, amikor a bogarak megeszik a játékost. És még egy: nem mindig következetesen üldözik a játékost a bogarak.


```

10 BORDER 4: PAPER 4: INK 0: CLS
15 LET l=0: LET z=0
100 LET l=l+1
105 LET e=INT (RND*200+100)
110 LET w=(RND>.8): LET we=INT (RND*e): LET s=(RND<.
3): LET se=INT (RND*e): LET r=(RND>.5): LET re=INT (R
ND*e)
114 REM A PALYA KIRAJZOLASA
115 CLS : PRINT l; ". lyuk", "tavolsag "; e
120 DRAW 120,80: PLOT 255,0: DRAW -120,80
125 PLOT 120,80: DRAW 16,0,PI/5: DRAW -16,0,PI/5
130 PLOT 128,81: DRAW 0,5: DRAW 2,-1: DRAW -2,-1
135 FOR i=0 TO PI STEP .2: PLOT i*20+93,85+5IN i*10:
DRAW 0,-10: NEXT i
141 LET z=z+1
145 IF s THEN INK 5: PLOT 180,60: DRAW 50,0,PI/10:
DRAW -10,20,PI: DRAW -40,-20,PI/5: INK 0
150 IF r THEN FOR i=1 TO 20 STEP 4: PLOT 140+i,70:
DRAW 3,-15: NEXT i: FOR i=20 TO 1 STEP -4: PLOT 140+i
,70: DRAW -3,-15: NEXT i
155 CIRCLE 124,10,10
160 INPUT "utes erosseg (1-20) ?";s: LET s=INT ABS s
: IF s<=0 OR s>20 THEN GO TO 160
165 LET g=INT (RND*15*s+s)
170 PRINT AT 1,0;"utes erosseg ";s
171 REM A GOLYD LOKESE
175 FOR i=-15 TO 0: BEEP .01,i: NEXT i: BEEP .04,10
180 CIRCLE INVERSE 1;124,10,10
185 FOR i=2 TO s: BEEP .04,i: PRINT AT 20-i,15;("o"
AND i<10);("o" AND i>=10): PRINT AT 21-i,15;" ": NEXT
i
190 FOR i=20-s TO g/15: BEEP .04,i: PRINT AT i,15;"o
": PRINT AT i-1,15;" ": NEXT i
193 REM FELTETELVIZSALAT
195 IF w AND ABS (e-we)<20 THEN GO TO 210
196 IF r AND ABS (e-re)<20 THEN GO TO 250
197 IF s AND ABS (s-se)<10 THEN GO TO 300
200 LET e=INT ABS (e-g): IF e>15 THEN GO TO 115
205 GO TO 350

```

```

207 REM A GOLYO AZ ERDOBEN
210 CLS : PRINT "A golyo az erdobe repult"
212 LET z=z+1
215 PLOT 180,20: DRAW 0,100: DRAW -40,-5: DRAW 45,35
: DRAW -30,-5: DRAW 35,20: DRAW INVERSE 1;10,0: DRAW
35,-20: DRAW -30,5: DRAW 45,-35: DRAW -40,5
220 DRAW 0,-100: PRINT AT 8,14;"FA-->"
225 INPUT "utes erosseg (1-20) ? ";s: LET s=INT ABS
s: IF s<=0 OR s>20 THEN GO TO 225
230 LET s=s*RND
235 FOR i=-15 TO 0: BEEP .01,i: NEXT i: BEEP .04,10
240 GO TO 200
245 REM A GOLYO AZ AROKBAN
250 CLS : PRINT "A golyo arokbaesett!"
255 FOR i=20 TO 150 STEP 20: PLOT.RND*10,i: FOR j=1
TO 6: DRAW 20,10: DRAW 20,-10: NEXT j: NEXT i
260 FOR i=0 TO 255 STEP 20: PLOT i,20: DRAW 100-i*.8
,130: NEXT i
265 LET z=z+1: GO TO 225
299 REM A GOLYO A TOBAN
300 CLS : PRINT "A golyo a toba esett. Elolrol kez
dheted ezt a lyukat"
305 FOR i=1 TO 5: BEEP .5,-10: NEXT i: PAUSE 100: GO
TO 105
350 IF e<2 THEN GO TO 400
352 CLS
353 PRINT "tavolsag: ";e
355 PLOT 20,70: DRAW 180,0,PI/2: DRAW -180,0,PI/2
360 PLOT 110,70: DRAW 0,30: DRAW 30,-10: DRAW -30,-1
0
365 CIRCLE 110+e*8,60,3
370 INPUT "lyuk (1-20) ? ";s: LET s=INT ABS s: IF s
=0 OR s>20 THEN GO TO 370
372 LET z=z+1
375 LET g=s*RND: GO TO 200
399 REM ERTEKELES
400 CLS : PRINT FLASH 1; BRIGHT 1;"Talalat"
405 FOR i=1 TO 3: FOR j=-10 TO 10: BEEP .04,j: NEXT
j: NEXT i


```

```
410 PRINT "Eredmeny: ";z;" Pont"  
415 IF 1<18 THEN PAUSE 200: GO TO 100  
420 PRINT " PAPER 2;"Es ez az on vegpontszama!"  
425 PRINT "'Akar meg játszani (i/n) ?"  
430 PAUSE 0: IF INKEY$="i" THEN RUN  
435 CLS : PRINT "Akkor viszontlatasra!"
```

Megjegyzés: Elég jól kihasználja a hang- és grafikai hatásokat a program, bár a „golyó a tóba esett” kiírás nem mondható grafikus hatásúnak. A játékos számára nem követhető sem az akadályok kikerülési módja (ha egyáltalán van ilyen), sem pedig a lyuk távolsága és az ütés erőssége közötti összefüggést. Az alapötlet viszont jó, érdemes saját fantázia szerint bővíteni! Próbáljuk meg pl. az ütés irányát a játékosra bízni! Figyeljük meg, hogy az egyszerű REM-kiírások a programlistában milyen nagy mértékben segítik a tájékozódást!

BOMBÁZÓ

Készítette: Farkas Judit



Leírás: Egy repülőgép halad balról jobbra a képernyőn, egy képzeletbeli város fölött. Előre megadható (bár ezen belül véletlenszerűen változik) a házak magassága és távolsága. A repülőgép, miután a kép jobb szélén elhagyta a várost, a bal oldalon egy karakterrel mélyebben újra megjelenik. A pilóta a B gomb segítségével szórhatja a bombákat. Ha a repülőgép háznak ütközik, vége a játéknak. Az a cél, hogy minél tovább, minél mélyebbre jussunk el a repülővel. Vízszintes irányban az M billentyű segítségével rakétákat is ki lehet lőni, de a kezdeti öt rakéta, ha nem takarékoskodunk, elfogy.

```

1>DATA 0,0,0,16,16,56,68,146,0,0,56,16,56,124,254
2 DATA 132,96,34,130,206,158,238,170,40,56,16,56,6
8,68,40,16
3 DATA 0,0,16,72,63,72,16,0,0,112,120,127,127,32,6
4,0,0,0,240,154,254,250,80,136
4 DATA 28,93,42,28,8,8,20,34,28,28,8,28,42,73,20,3
4: LET hs=0
5 GO SUB 3000
8 BORDER 1: PAPER 7: INK 0: CLS
9 LET b=5: LET sc=0
10 REM ** 1-9-IG UDG-k**
15 REM **ERTEKADAS**
25 INPUT "Hazak magassaga? (1-10)";k: IF k<1 OR k>1
0 THEN GO TO 25
26 INPUT "Hazak kozti tavolsag? (0-3)";r: LET r=r+1
: IF r<1 OR r>4 THEN GO TO 26
27 LET k=k+2: BRIGHT 1
29 REM **VAROS KIRAJZOLASA**
30 FOR f=4 TO 28 STEP r
40 LET l=RND*10+k: FOR g=21 TO 1 STEP -1: PRINT AT
g,f;"███": NEXT g: PRINT AT g,f;CHR$(145+RND*1)
50 NEXT f
51 FOR x=0 TO 31: PRINT AT 21,x; INK 2;"███": NEXT x
52 BRIGHT 0
55 LET y=1
57 INK 0: PRINT AT 0,0; BRIGHT 0;"PONT: ";sc;AT 0,2
2;b;" RAKETA ";AT 0,10;"CSUCS: ";hs
58 INK 1: POKE 22562,56
60 FOR x=0 TO 30
64 PRINT AT y,x;"███": IF y=20 AND x=30 THEN GO T
0 2500
65 REM **IRANYITAS**
66 LET l$=INKEY$: IF l$<>" " THEN GO TO 500
67 GO SUB 1000
70 PRINT AT y,x;" "
80 NEXT x
85 PRINT AT y,x;" "
90 LET y=y+1: GO TO 60
500 IF l$="b" THEN GO TO 600

```

```

510 IF I#="m" THEN GO TO 520
515 GO TO 70
517 REM **RAKETA**
520 LET b=b-1: IF b<=-1 THEN LET b=0: GO TO 67
522 PRINT AT 0,22;b;" RAKETA"
523 FOR m=x+2 TO x+8
525 IF m>=31 THEN GO TO 70
526 LET j=ATTR (y,m): IF j<>56 THEN PRINT AT y,m;"
": GO TO 70
527 PRINT AT y,m; INK 1;"@-": BEEP .05,5: PRINT AT y
,m;" ": NEXT m
540 GO TO 64
590 REM **BOMBA**
600 LET e=x: LET f=y+1
610 IF ATTR (f,e)<>56 THEN GO TO 700
620 GO SUB 2000
630 IF f=21 THEN GO TO 70
640 GO TO 610
700 FOR g=1 TO 6
710 GO SUB 2000
711 IF f=21 THEN GO TO 70
712 LET sc=sc+1: PRINT AT 0,22;b;" RAKETA"
713 REM **PONTSZAMLALAS**
714 PRINT AT 0,0;"PONT: ";sc
715 NEXT g
716 PRINT AT f-1,e; BRIGHT 1; INK 2;"@#!"
720 GO TO 70
990 REM **UTKOZES**
1000 LET l=ATTR (y,x+2): IF l=56 OR l=57 OR l=58 THEN
RETURN
1010 RESTORE 73: FOR f=1 TO 11: READ a,b: BEEP a,b: N
EXT f: DATA .35,-1,.3,-1,.1,-1,.4,-1,.25,2,.15,1,.25,
1,.15,-1,.25,-1,.15,-2,.25,-1
1011 FOR f=1 TO 3: FOR g=0 TO 7: INK g: PRINT AT y,x;
"#####": BEEP .3-(f/10),-g-10: NEXT g: NEXT f
1015 REM **ERTEKELES**
1020 INK 6: PAPER 0: CLS : PRINT " A TE PONTJAID S
ZAMA ";sc;".": IF sc>hs THEN GO TO 4000
1030 PRINT "' ' ' UJRAKEZDESHEZ NYOMJ!": PAUSE 0: G

```

```


0 TO 8
2000 PRINT AT y,x;"  ": LET x=x+1: PRINT AT y,x;"  "
": IF x=31 THEN PRINT AT y,x;"  ": LET x=0: LET y=y+
1
2002 GO SUB 1000
2005 IF f=21 THEN RETURN
2010 PRINT AT f,e;"  ": BEEP .02,1: PRINT AT f,e; INK
0;"  ": LET f=f+1
2030 RETURN
2500 PRINT AT 20,30;"  ": PRINT AT 20,0;"  "
2510 FOR f=1 TO 150: PRINT AT 20,4;"  ";AT 20,4;"  ":
NEXT f: BORDER 1
2520 FOR f=1 TO 2: BEEP .2,1: BEEP .1,1: BEEP .15,1:
BEEP .25,5: BEEP .1,1: BEEP .4,5: PAUSE 25: NEXT f
2530 CLS : PRINT AT 0,1; FLASH 1; INK 6; PAPER 0;"  G
RATULALOK, LESZALT 1"
2540 LET y=21: FOR f=1 TO 30: BORDER RND+6: LET y=y-2
1/30: PRINT AT y,f;"  ": BEEP (31-f)/300,-10: PRINT
AT y,f;"  ": NEXT f: PRINT AT y,f;"  "
2541 FOR f=0 TO 30: PRINT AT 0,f;"  ": BEEP .05,10:
PRINT AT 0,f;"  ": NEXT f
2545 INK 0: PAPER 7
2550 CLS
2555 LET k=k+1: IF k=0 THEN LET k=1
2556 LET b=5
2560 BRIGHT 1: GO TO 30
2990 REM **UDG POKOLASA**
3000 FOR f=0 TO 7 STEP 2: POKE USR "  "+f,255: POKE U
SR "  "+f+1,170: NEXT f
3002 FOR f=0 TO 71: READ a: POKE USR "  "+f,a: NEXT f
3004 REM **JATEK ELEJE**
3005 LET a$="  "
3006 LET b$="B O M B A Z O"
3009 BORDER 1: INK 6: PAPER 0: CLS
3010 FOR f=9 TO 11: PRINT AT f,9;a$: NEXT f
3020 FOR s=0 TO 13: PRINT AT s,0;"  "
3025 IF s=9 OR s=11 THEN BEEP .5,5: FOR f=2 TO 31: P
RINT AT s,f;"  ": BEEP .02,5: PRINT AT s,f;"  ": NEXT
f

```


Megjegyzés: Barátságos a bemutatkozás és az értékelés, bár a „FOLYTATÁSHOZ NYOMJ” felirat ügyetlenül hangzik. Nagyon hiányzik az útmutatás, legalább annyi, hogy melyik gombbal lehet bombázni és lőni. Bár a program alapötlete nem új, jól megírt, használható játék. Lehet tökéletesíteni, a REM-ek ebben segítenek. Kihasználja a program a gép hanghatásait is. Vigyázat, bombázáskor a túl magas házakat egy bombatalálattal nem lehet lerombolni! „Leszálláskor” kellemes meglepetés, hogy folytatni lehet a játékot, folytatólagos pontszerzéssel. Megtévesztő azonban, hogy a házak magasságának szabályozása úgy adható meg, hogy 10 a legalacsonyabb, 1 a legmagasabb értéket jelöli. És még valami: ezeknél az ismételt (leszállás utáni) játékoknál mintha a házak magassága egyre csökkenne (RND-probléma?).

PIROS HÁZAK

Készítette: Radácsy Tivadar



Leírás: Piros háztetők végtelen sorát kell kerülgetni. A játékos figurája normál állapotban mindig balra „húz”, tetszőleges billentyű lenyomásával viszont jobbra mozgatható. A vízszintes táblák a játékost felfelé mozgatják, és ha eléri a képernyő tetejét, a program véget ér, és közli az elért pontszámot, valamint az addig legmagasabb rekordot.

```
4 REM UDG pokolasa
5 DATA 0,8,8,8,255,255,126,60,170,85,170,85,170,85
,170,85
6 FOR i=0 TO 15: READ a: POKE USR "↓"+i,a: NEXT i
7 LET h=1
```

```


8 POKE 23692,-1: LET s=0: LET u=15: LET p=10
9 REM foprogram
10 PRINT AT u-1,p;" "
20 LET p=p-(p>1)
30 LET p=p+(INKEY$("<>"))*2*(p<20)
40 LET n=ATTR (u,p)
50 IF n=58 THEN LET u=u-1: BEEP .1,u
54 REM kirajzolas
55 IF u=0 THEN GO TO 100
60 PRINT AT u,p;" "
70 PRINT AT 21,0;
80 PRINT TAB RND*20; INK 2;" ██████████": PRINT
85 FOR j=1 TO 10: NEXT j
87 REM pontozas
90 LET s=s+u: GO TO 10
95 REM jatek vege
100 IF h<s THEN LET h=s
110 PRINT AT 10,0; INK 1;"ELERT PONT: ";s;" REKORD: ";h
120 FOR i=0 TO 100: BEEP .002,2: NEXT i
130 PAUSE 0: CLS : GO TO 8

```

Megjegyzés: Egyik változata a „meteoros” játéknak (l. 12. lista), de a piros házakkal való ütközést az ATTR függvénnyel vizsgálhatja. Nem mulasztja el a programozó a zenekíséretet sem. Egészen kezdőknek is jól követhető, mert rövid, és a megjegyzések útbaigazítanak.

AUTÓ

Készítette: Radácsy Tivadar



Leírás: Az autót egy kezdeti egyenes útszakasz után véletlenszerű cikcakkokban haladó úton kell vezetni úgy, hogy az időnként felbukkanó csatornatezőkre nem szabad ráhajtani. Q betűvel (balra), P betűvel (jobbra) lehet az autót irányítani. Minél tovább jutunk, annál sikeresebb a játék.

```
1>REM UDG pokolasa
10 DATA 170,85,170,85,170,85,170,85: FOR i=0 TO 7:
READ a: POKE USR "███"+i,a: NEXT i
15 REM ertekadas
30 LET c=13: LET a=16
```

```


35 REM a kezdopalya
40 FOR i=0 TO 21: PRINT AT i,0;"
      " : NEXT i
50 PRINT AT 10,2;" KEZDESHEZ"; AT 10,22;"
NYOMJ! "
60 REM foprogram
70 PRINT AT 2,a;"V"
95 REM irányítás
100 LET a=a+(INKEY$="p")-(INKEY$="q")
120 LET c=(c+INT (RND*5)-2)
140 LET c=c-(2*(c>22))+(1*(c<2))
150 REM allando palya
160 PRINT AT 21,1;"
170 PRINT AT 21,c;"
180 GO TO 190+(60*(RND>.92))
190 PRINT AT 21,31;" : POKE 23692,255
195 REM utkozes figyelese
200 GO TO 70+((ATTR (2,a)<>57)*140)
205 REM ukozes, jatek vege
210 PRINT AT 2,a;"*"
220 FOR i=5 TO 0 STEP -.1: BEEP .002,i: BEEP .002,i:
NEXT i
230 PAUSE 0: RUN
245 REM csatornafedel
250 PRINT AT 21,c+4;" : GO TO 190

```

Megjegyzés: Szintén a géppel való ismerkedésre alkalmas programlista. A játék fogyatékosága leginkább az értékelés hiánya. Sem pontot, sem egyéb visszajelzést nem kap a játékos az elért eredményről. Lehet tehát módosítani és kiegészíteni az alapötletet, az eligazodást segítő REM-ek alapján.

KRÁTER

Készítette: Radácsy Tivadar



Leírás: Az irányítható űrhajóval kell leszállni a Hold krátereibe. Bármelyik kráterbe le lehet szállni, de vigyázni kell, nehogy a falnak ütközzön az űrhajó bármelyik része. Az üzemanyaggal is gazdálkodni kell, mert ha elfogy, akkor az űrhajó kormányozhatatlanná válik és lezuhan. A kapott pontszám a nehézségi foktól függő szorzószámtól és a megtakarított üzemanyagtól függően változik.

Az űrhajó irányítása:

- O – balra
- P – jobbra
- Q – felfelé
- A – lefelé

```

10 REM (C) 1985 t
15 REM **ERTEKADAS**
20 LET hs=0
25 LET l=0
30 GO SUB 900: GO TO 800
40 INK 7
45 REM **AZ URHAJO**
50 LET a$=" :███: "
60 LET b$=" ████ "
70 LET c$="  "
75 REM **AZ URHAJO VALTOZOI**
80 LET x=INT (RND*20)+1: LET y=1: LET dx=0: LET dy=
0: LET s=255
85 REM **IRANYITAS**
90 IF ATTR (y+3,x+1)=14 THEN GO TO 580
100 LET dx=dx+(dx<1)*(INKEY$="p")-(dx>-1)*(INKEY$="o
")
110 LET dy=dy+(dy<1)*(INKEY$="a")-(dy>-1)*(INKEY$="q
")
115 REM **A CSILLAGOK**
120 INK 7: PLOT 64,40: PLOT 240,160: PLOT 100,100: P
LOT 200,140: PLOT 60,80: PLOT 130,85: PLOT 80,130: PL
OT 150,150: PLOT 225,60: PLOT 180,110: PLOT 130,120:
PLOT 19,170: PLOT 115,160
125 REM **ELMOZDULAS**
130 LET a=x: LET b=y
140 LET x=x+dx: LET x=x-(x>28)+(x<2): LET y=y+dy: LE
T y=y-(y>17)+(y<2)
145 REM **UTKOZES FIGYELES**
150 FOR n=0 TO 2: FOR m=0 TO 2: IF ATTR (y+n,x+m)=8
OR ATTR (y+n,x+m)=15 THEN GO TO 530
160 NEXT m: NEXT n
165 REM **REGI POZICIO TORLES**
170 IF dx=1 THEN PRINT AT y,a;" ";AT y+1,a;" ";AT y
+2,a;" "
180 IF dx=-1 THEN PRINT AT y,a+2;" ";AT y+1,a+2;" "
;AT y+2,a+2;" "
190 IF dy=1 THEN PRINT AT b,x;" "
200 IF dy=-1 THEN PRINT AT b+2,x;" "

```



```

450 PRINT AT 11,0;" ";AT 11,29;" "
460 PRINT AT 10,0;" ";AT 10,28;" ";AT 11,1
1;" ";AT 11,21;" "
470 PRINT AT 9,0;" ";AT 9,27;" "
480 PRINT AT 8,0;" ";AT 8,27;" "
490 PRINT AT 7,0;" ";AT 7,27;" "
500 PRINT AT 6,28;" ";AT 5,29;" ";AT 4,30;" "
";AT 3,31;" ";AT 2,31;" "
510 PAPER 0
520 RETURN
530 REM **ROBBANAS**
535 PRINT AT y-1,x;" ";AT y,x-1;"BANG!";AT y+1,x
;" "
540 FOR I=1 TO 15: BEEP .05,40: BEEP .05,30: NEXT I
550 PRINT AT 1,11;"JATEK VEGE"
560 PRINT AT 4,1;"NYDMJ"
570 GO TO 670
590 REM **SIKERES LESZALLAS**
600 PRINT AT 1,8;"GRATULALOK!"
610 PRINT AT 4,1;"PONT: "
620 LET t=2: IF x<12 THEN LET t=5
630 IF x>23 THEN LET t=50
640 LET ll=(s-55)
650 PRINT 1;"+";ll;"x";t;" = ";: LET l=1+ll*t: PRINT
l
660 FOR n=1 TO 10: BEEP .05,1: BEEP .05,5: BEEP .05,
8: BEEP .05,13: BEEP .05,8: BEEP .05,5: NEXT n: BEEP
.06,1
665 CLS : GO SUB 290: GO TO 40
669 REM **JATEK VEGE**
670 IF INKEY$<>" " THEN GO TO 670
680 IF INKEY$="" THEN GO TO 680
690 CLS
700 IF l>hs THEN PRINT ""UJ CSUCS!!!"
710 IF l>hs THEN LET hs=l
720 PRINT ""A REKORD : ";hs;"A FONTOD: ";l
725 FOR i=0 TO 100: NEXT i: PAUSE 0
730 IF INKEY$<>" " THEN GO TO 730
740 GO TO 840

```

```

750 FOR i=0 TO 2
760 LET a=ATTR (y+3,x+i): IF a<>7 THEN GO TO 530
770 NEXT i
780 PRINT AT y+1,x;a$;AT y+2,x;b$;AT y+3,x;c$;AT y,x
;"  ": LET y=y+1: BEEP .02,40-y
790 GO TO 750
800 BORDER 0: PAPER 0: INK 7: CLS
805 REM **INFORMACIOK**
810 PRINT TAB 9;"Instrukciok": PRINT TAB 9;"-----
---"
812 PRINT AT 4,0;" Szallj le az urhajoval az hold- b
eli tajon! O-val balra          p-vel jobb
ra          O-val fel
A-val lefele ira- nyithatod az urhajodat."
815 PRINT AT 20,0;" Leszallashoz felkészültel?
(Ha igen, nyomj egy gombot)"
820 IF INKEY$="" THEN GO TO 820
825 GO SUB 290
830 GO TO 40
860 GO TO 25
895 REM **UDG-k**
900 DATA 0,0,3,7,23,15,15,23,0,126,255,255,255,255,2
55,255,0,0,192,224,232,240,240,232,7,3,3,7,7,7,15  3
910 DATA 224,192,192,224,224,224,240,232,22,20,60,32
,96,64,64,224,60,60,60,0,0,0,0,0,104,40,60,4,6,2,2,7
920 DATA 255,24,255,255,255,255,255,255,62,62,62,62,
255,255,63,63,63,63,62,54,54,54,119,119,1,1,1,1,1,1,1
,1
930 DATA 255,127,31,7,3,3,1,1,0,128,224,248,252,252,
254,254,255,254,248,224,192,192,128,128,0,1,7,31,63,6
3,127,127
960 RESTORE 900
970 FOR i=0 TO 127: READ a: POKE USR " " + i, a: NEXT
i
980 RETURN

```


UDG



Megjegyzés: A játék bevezetője egy tájékoztató, igen, így kéne másutt is. A „NYOMJ” ugyan udvariatlan, (bár az angol „PRESS” szó magyar megfelelője). A játék maga kidolgozott, nem maradtak el sem a hanghatások, se a REM-ek.

TÁLTOS A KAKTUSZOK KÖZÖTT

Készítette: Radácsy Tivadar



Leírás: A képernyőn egy sivatagi táj jelenik meg, kétoldalt egy-egy cowboy. A kaktuszok és a ló felfogják a lövéseket. A játéknak ebben a változatban soha nincs vége. A pontszám úgy alakul, hogy ha a játékos lövi ki az ellenfelet, akkor eggyel nő, ha a játékost találja el a másik cowboy, akkor eggyel csökken a pontszám. Irányítás:

Q – felfelé

A – lefelé

P – lövés

A program irányítása az UDG-k (programozott karakterek) miatt RUN 500 utasítással történik.

```
10 REM Ertekadasok
```

```
20 GO SUB 380
```

```
30 BORDER 6: PAPER 6: INK 1: LET s=0: LET h5=0
```

```
35 REM Kirajzolas
```

```

40 CLS : PRINT PAPER 1; INK 7;" PONT   : ";s;"
   REKORD   : ";hs
50 FOR i=1 TO 4: LET a=i*3+INT (RND*3): LET b=i+2+I
NT (RND*6): IF RND>.5 THEN LET b=b+14
60 PRINT OVER 1; INK 4;AT a,b;" ■";AT a+1,b;" ■"
";AT a+2,b;" ■ "
70 NEXT i
80 LET a=INT (RND*18)+2: PRINT INK 0;AT a,14;" ■"
";AT a+1,14;" ■"
90 LET y1=4: LET y2=10
95 REM Foprogram
100 LET y2=y2+((RND>.5) OR (y1>y2))-((RND>.5) OR (y1
<y2)): IF y2>=20 THEN LET y2=19
110 PRINT INK 0;AT y1-1,1;" ";AT y1+2,1;" ";AT y2
-1,29;" ";AT y2+2,29;" "; INK 2;AT y1,1;" ─";AT y
1+1,1;" ┆ ";AT y2,29;" ─";AT y2+1,29;" ┆ "
120 LET y1=y1+(INKEY$="a")*(y1<19)-(INKEY$="q")*(y1>
2)
130 IF INKEY$="p" THEN GO SUB 160
140 IF y2=y1 OR y2=y1+1 AND RND>.3 THEN GO SUB 270
150 GO TO 100
155 REM A te lovesed
160 FOR i=3 TO 29: IF ATTR (y1,i)<>49 THEN GO TO 19
0
170 PRINT AT y1,i;" =": BEEP .01,20: PRINT AT y1,i;"
"
180 NEXT i
190 IF ATTR (y1,i)=50 THEN GO TO 220
200 LET k=4: IF ATTR (y1,i)=48 THEN LET k=0
205 REM A halal
210 FOR n=0 TO 6: PRINT OVER 1; INK n;AT y1,i;"*":
BEEP .05,n: NEXT n: PRINT OVER 1; INK k; PAPER 6;AT
y1,i;"*": RETURN
220 PRINT INK 2;AT y2,29;" ";AT y2+1,29;" ─"
230 PAUSE 20: FOR i=1 TO 10: BEEP .02,30: BEEP .02,4
0: NEXT i
240 RESTORE : FOR n=1 TO 11: READ a,b: PAUSE 1: BEEP
a,b: NEXT n

```

```

250 LET s=s+1: IF s>hs THEN LET hs=s
260 GO TO 40
265 REM Az o lovese
270 FOR i=28 TO 2 STEP -1: IF ATTR (y2,i)<>49 THEN
GO TO 300
280 PRINT AT y2,i;"-": BEEP .01,30: PRINT AT y2,i;"
"
290 NEXT i
300 IF ATTR (y2,i)=50 THEN GO TO 330
310 LET k=4: IF ATTR (y1,i)=48 THEN LET k=0
315 REM A Halal II.
320 FOR n=0 TO 6: PRINT OVER 1; INK n;AT y2,i;"*":
BEEP .05,n: NEXT n: PRINT OVER 1; INK k; PAPER 6;AT
y2,i;"*": RETURN
330 PRINT INK 2;AT y1,0;" ";AT y1+1,0;" -----"
340 PAUSE 20: FOR i=1 TO 10: BEEP .02,10: BEEP .02,2
0: NEXT i
350 RESTORE : FOR n=1 TO 11: READ a,b: PAUSE 1: BEEP
a,b: NEXT n
360 LET s=s+1: GO TO 40
365 REM Gyaszdulo
370 DATA .66,0,.66,0,.2,0,.66,0,.46,3,.2,2,.46,2,.2,
0,.46,0,.2,-1,.66,0
375 REM Informaciok
380 GO SUB 500: CLS : PRINT "Taltos lo a kaktuszok k
ozott.";AT 2,10;"INFORMACIOK";AT 3,10;"-----";A
T 6,0;" Te a jatekban a baloldali cow-boy vagy. Le k
ell lonod a masik oldalon allo ellenseget. A koz- tet
ek levo akadalyok felfogjak a lovedeket!"
390 PRINT AT 12,0;" A jatek addig tart, amig meg n
em unod."
400 PRINT AT 15,0;" A pontod ha te lovod le ot, a
kkor 1-el no, ha o lo le te- ged, akkor 1-el csokke
n. A csucs legmagasabb erteket orzi."
410 PRINT AT 21,13;"NYOMJ!"
450 PAUSE 0: RETURN
495 REM UDG-k
500 DATA 12,30,63,63,63,63,63,63,0,0,0,0,0,24,60,60,

```

```

63,63,63,63,31,15,7,7,60,60,60,124,252,248,240,224
510 DATA 7,7,7,7,7,7,7,3,192,192,192,192,192,192,192
,128,28,62,28,28,24,63,63,60,0,0,0,0,8,254,248,16
520 DATA 60,60,28,24,24,24,24,28,0,0,0,0,16,127,31,8
,56,124,56,56,24,252,252,60,60,60,56,24,24,24,24,56
530 DATA 0,0,0,0,135,255,255,3,0,0,0,2,239,255,255,2
26,0,0,0,0,60,60,0,0,0,16,22,15,31,63,59,51
540 DATA 0,0,0,0,128,227,255,255,0,0,0,0,224,240,252
,247,3,3,7,14,26,18,20,20,255,254,192,0,0,0,0,0
550 DATA 243,121,72,108,20,18,34,102
560 RESTORE 500: FOR i=0 TO 167: READ a: POKE USR " #
"+i,a: NEXT i
570 RETURN


```

UDG

Megjegyzés: Kezdjük a címmel. Jobb lett volna, ha a táltos helyett egy élet-szerű paripa kerül a képernyőre, amelyik nem állja a lövést. A golyózápport egyébként a kaktuszok sem bírják. Könnyen kiküszöbölhető lett volna a bonyodalmas RUN 500 indítás. Olykor a halott cowboy felett még megmarad valami kísértetként is. Ennyi kritika után mondhatnánk, hogy a program inkább az az állatorvosi „beteg ló” a kaktuszok közt, ami jól példázza a játék-program-betegségeket.

SÍELŐ

Készítette: Radácsy Tivadar



Leírás: Fák és zászlók között haladó síversenyzőt lát a játékos, akit a felfelé haladó akadályok között a nyílombokkal lehet jobbra, ill. balra irányítani. Fának ütközve a figura „tolókocsi”-ra változik, és egy figyelmeztető kiírás jelenik meg a képernyőn.

```
10 REM **ERTEKADAS**  
20 BORDER 7: PAPER 7: INK 0: CLS  
40 CLS : LET X=15: LET Y=11: LET F=0: LET SC=0: LET  
S=10  
50 PRINT AT 21,0;: GO TO 110  
55 REM **IRANYITAS**
```

```

60 LET P=X: LET X=X+(INKEY$="8")*(X<29)-(INKEY$="5"
)*(X>1)
70 PRINT
75 REM **UTKOZES FIGYELESE**
80 IF ATTR (Y,X)<>56 OR ATTR (Y,X+1)<>56 OR ATTR (Y
+1,X)<>56 OR ATTR (Y+1,X+1)<>56 THEN GO SUB 270
85 REM **KIRAJZOLAS**
90 PRINT AT Y-1,P;" ";AT Y,P;" ";AT Y,X;" ";AT
Y+1,X;" "
100 BEEP .01,35: PRINT AT 21,0;: RETURN
105 REM **PALYA KIRAJZOLASA**
110 POKE 23692,0
120 LET T=INT (RND*18): LET V=4+INT (RND*10): PRINT
INK 4;AT 21,T;" ";AT 21,T+V;" ": GO SUB 60
130 PRINT INK 4;AT 21,T;" ";AT 21,T+V;" ": GO
SUB 60
140 PRINT INK 4;AT 21,T;" ";AT 21,T+V;" ": GO
SUB 60
150 PRINT INK 4;AT 21,T;" ";AT 21,T+V;" ": GO
SUB 60
160 PRINT : GO SUB 60
170 PRINT : GO SUB 60
180 IF X>S AND X<S+9 THEN LET SC=SC+(F>0)
190 PRINT : GO SUB 60
200 PRINT : GO SUB 60
210 LET S=5+INT (RND*16): PRINT INK 1;AT 21,S;" ";
INK 2;AT 21,S+10;" ": GO SUB 60
220 PRINT INK 1;AT 21,S;" "; INK 2;AT 21,S+10;" "
: GO SUB 60
230 FOR I=1 TO 4: PRINT : GO SUB 60
240 NEXT I
250 LET F=F+1
260 GO TO 110
265 REM **UTKOZES FIGYELES**
270 IF ATTR (Y,X)=57 OR ATTR (Y,X+1)=57 OR ATTR (Y+1
,X)=57 OR ATTR (Y+1,X+1)=57 THEN LET C=1: GO TO 360
280 IF ATTR (Y,X)=58 OR ATTR (Y,X+1)=58 OR ATTR (Y+1
,X)=58 OR ATTR (Y+1,X+1)=58 THEN LET C=2: GO TO 360
290 FOR I=1 TO 10: BEEP .1,-10: BEEP .1,-25: NEXT I:

```

```

PRINT INK 2; FLASH 1; AT Y-1, P; "  "; AT Y, P; "  ": F
OR I=60 TO -30 STEP -1: BEEP .01, I: NEXT I
295 REM **JATEK VEGE**
300 PRINT INK 2; AT Y-1, P; "  "; AT Y, P; "  "; AT 0, 0;
FLASH 1; "FANAK MENDEL, HAVER!"; AT 2, 4; "PROBALD MEG M
EGEGYSZER!"
310 PRINT FLASH 1; AT 21, 1; "UJ PROBALKOZASHOZ NYOMJ!
"
320 IF F<>0 THEN PRINT AT 10, 0; "ELERT TELJESITMENY:
"; SC; " Z. / "; F; " FA."
330 FOR I=1 TO 50: NEXT I
340 IF INKEY#="" THEN GO TO 340
350 GO TO 40
360 PRINT AT Y-1, X-1; INK C; "  "
370 PRINT AT Y, -1; "  "; AT Y+1, X-1; "  "; AT Y+2, X-
1, "  ": LET Y=Y+1
380 IF Y=20 THEN LET Y=Y-1
390 RETURN
495 REM **UDG-k**
500 DATA 1, 3, 3, 1, 3, 15, 19, 43, 128, 192, 192, 128, 192, 240,
200, 212, 11, 11, 14, 12, 12, 8, 8, 8, 208, 208, 112, 48, 48, 16, 16,
16
510 DATA 3, 15, 63, 255, 63, 15, 3, 1, 1, 1, 1, 1, 1, 1, 1, 1, 3, 7
, 15, 31, 63, 127, 255, 128, 192, 224, 240, 248, 252, 254, 255
520 DATA 3, 3, 3, 3, 0, 0, 0, 0, 192, 192, 192, 192, 0, 0, 0, 0, 0, 6
4, 32, 32, 112, 240, 251, 252, 0, 0, 0, 96, 144, 136, 8, 7
530 DATA 60, 126, 126, 60, 31, 60, 124, 252, 3, 7, 7, 5, 13, 36, 5
4, 6, 132, 28, 34, 85, 73, 213, 34, 28
550 FOR I=0 TO 119: READ A: POKE USR "  "+I, A: NEXT
I: RUN

```


UDG

Megjegyzés: Itt ismét hiányzik a játékost tájékoztató kiírás. A program RUN 500 utasítással indul, az UDG-k (programozható karakterek) miatt. Néhány szépséghiba előfordul ebben a játékban is. A „találatot” (fának ütközést)

jelző kiírásban pl. a „FA” szó F betűje az egyik sor végén, az A betűje a következő sor elején van. Mulatságos ötlet viszont baleset után az átváltás a tolokocsis figurára. Néhány REM itt is lehetőséget ad a továbbfejlesztésre. A látvány esetleg bővebb zenekíséretet is megérdemel, és egyéb, pl. az időtől függő, célba érkezésig tartó értékelést is be lehetne építeni.

LABIRINTUS

Készítette: Radácsy Tivadar



Leírás: A szokásos labirintusfeladat saját készítésű változata. Betöltés után meg lehet adni a nehézségi szintet (1–10), a program ennek megfelelően egy labirintustérképet generál. Ez bármikor, de összesen csak háromszor a T betű lenyomásával hívható. A képernyőn az égtájaknak megfelelően számok mutatják az irányítást. A program a helyes út megtalálásakor zenés gratuláció után megkérdezi, hogy kíván-e a játékos még egyszer játszani, és ha igen, újraindul.

```

1> RESTORE 1:FOR n=0TO 18:READ m:POKE USR " " +n,m
:NEXT n:DATA 33,0,88,6,24,14,32,52,13,35,194,95,127,5
,194,93,12,7,201
5 LET tk=3: PAPER 1: INK 7: BORDER 1: CLS : DIM m
$(704): LET fal=0
6 INPUT "Nehezsegi szint (1-10) ";d: LET d=d/10
10 LET l=20: LET c=16: LET ld=-1: LET cd=0
20 PRINT AT 1,c;"■■■■"
25 LET m$(1*32+c+1)="■■■■"
30 LET l=l+ld: LET c=c+cd
45 IF RND>d THEN GO TO 120
50 IF INT (1/2)<>1/2 OR INT (c/2)<>c/2 THEN GO TO
120
60 LET r=INT (RND*4)
70 LET ld=(r=2)-(r=3): LET cd=(r=0)-(r=1)
80 BEEP .01,RND*50
120 IF l=0 THEN GO TO 160
130 IF l+ld>20 THEN LET ld=-ld
140 IF c+cd<1 OR c+cd>30 THEN LET cd=-cd
150 GO TO 20
160 LET e=c
1000 RESTORE 1000: DIM d(6): FOR n=1 TO 6: READ d(n):
NEXT n: DATA 22,19,17,15,13,12
1010 RESTORE 1020: DIM p(6): DIM c(4,6,2): FOR m=1 TO
4: FOR n=1 TO 6: READ c(m,n,1): READ c(m,n,2): NEXT
n: NEXT m
1020 DATA 0,0,22,22,41,41,58,58,73,73,86,86
1030 DATA 0,174,22,152,41,133,58,116,73,101,86,88
1040 DATA 174,174,152,152,133,133,116,116,101,101,88,
88
1050 DATA 174,0,152,22,133,41,116,58,101,73,88,86
1060 LET l=20: LET c=16: LET ld=-1: LET cd=0
2000 FOR n=1 TO 6
2010 LET a=1*32+c+1+(32*n*ld)+(n*cd)
2020 IF (l+n>20 AND ld=1) OR (l-n<1 AND ld=-1) OR (c-
1<1 AND cd=-1) OR (c+1>31 AND cd=1) THEN GO TO 2030
2021 IF cd=1 AND ld=0 THEN LET p(n)=(m$(a)=" ")
2022 IF cd=-1 AND ld=0 THEN LET p(n)=(m$(a+32)="■■■■"
+2*(m$(a-32)="■■■■")+4*(m$(a)=" "))

```

```


2023 IF cd=0 AND ld=1 THEN LET p(n)=(m$(a+1)="████")+2
*(m$(a-1)="████")+4*(m$(a)=" ")
2024 IF cd=0 AND ld=-1 THEN LET p(n)=(m$(a-1)="████")+
2*(m$(a+1)="████")+4*(m$(a)=" ")
2025 IF p(n)>4 THEN LET p(n)=4: GO TO 2030
2026 NEXT n
2030 GO SUB 9000
2040 IF c=e AND l=1 THEN GO TO 3000
2045 PRINT AT 9,23;"IRANY:";AT 11,23;("ESZAK " AND ld
=-1 AND cd=0)+(" DEL " AND ld=1 AND cd=0)+("KELET "
AND ld=0 AND cd=1)+("NYUGAT" AND ld=0 AND cd=-1)
2059 PRINT AT 15,24;" ^";AT 16,24;" 7";AT 17,23;"<5 8
>";AT 19,22;"T=TERKEP"
2060 PAUSE 0: LET z$=INKEY$
2061 IF z$="t" AND tk>0 THEN BEEP .2,60: CLS : LET t
k=tk-1: PRINT m$: PRINT AT 1,c; FLASH 1; PAPER 2;"+";
INVERSE 1;AT 0,e;"E": PAUSE 0: CLS : GO TO 2000
2062 BEEP .02,RND*30
2065 IF z$="7" THEN LET l=1+ld: LET c=c+cd: LET p(1)
=p(2): LET p(2)=p(3): LET p(3)=p(4): LET p(4)=p(5): L
ET p(5)=p(6): FOR n=6 TO 6: GO TO 2010
2080 IF z$="5" AND ld=-1 AND cd=0 THEN LET ld=0: LET
cd=-1: GO TO 2000
2090 IF z$="5" AND ld=1 AND cd=0 THEN LET ld=0: LET
cd=1: GO TO 2000
2100 IF z$="5" AND ld=0 AND cd=-1 THEN LET ld=1: LET
cd=0: GO TO 2000
2110 IF z$="5" AND ld=0 AND cd=1 THEN LET ld=-1: LET
cd=0: GO TO 2000
2120 IF z$="8" AND ld=-1 AND cd=0 THEN LET ld=0: LET
cd=1: GO TO 2000
2130 IF z$="8" AND ld=1 AND cd=0 THEN LET ld=0: LET
cd=-1: GO TO 2000
2140 IF z$="8" AND ld=0 AND cd=1 THEN LET ld=1: LET
cd=0: GO TO 2000
2150 IF z$="8" AND ld=0 AND cd=-1 THEN LET ld=-1: LE
T cd=0: GO TO 2000
2160 GO TO 2060
3000 CLS : PRINT AT 10,8;" GRATULALOK ": FOR n=1 TO

```

```

300: BORDER RND*7: BEEP .002,n/5: NEXT n
3020 CLS : BORDER 1
3030 DIM a$(1): INPUT "Jatszol megegyszer? (i/n) "; L
INE a$: IF a$="1" THEN RUN
3040 IF a$="n" THEN STOP
3050 GO TO 3030
9000 CLS : PLOT 0,0: DRAW 0,174: DRAW 174,0: DRAW 0,-
174: DRAW -174,0: FOR n=1 TO 5
9010 FOR m=1 TO 4
9015 IF p(n)=4 THEN LET fal=n: GO TO 9060
9020 PLOT c(m,n,1),c(m,n,2)
9040 IF (m<3 AND (p(n)=0 OR p(n)=2)) OR (m>2 AND (p(n)
)=0 OR p(n)=1)) THEN DRAW c(m,n+1,1)-c(m,n,1),c(m,n+
1,2)-c(m,n,2)
9045 NEXT m
9050 NEXT n
9060 FOR n=1 TO 6
9065 IF fal=n THEN GO TO 9110
9066 IF n=6 THEN GO TO 9090
9070 IF p(n)=1 OR p(n)=3 THEN PLOT c(2,n,1),c(2,n,2)
: DRAW 0,c(1,n,2)-c(2,n,2): PLOT c(2,n+1,1)-d(n),c(2,
n+1,2): DRAW d(n),0: DRAW 0,c(1,n+1,2)-c(2,n+1,2): DR
AW -d(n),0
9080 IF p(n)=2 OR p(n)=3 THEN PLOT c(3,n,1),c(3,n,2)
: DRAW 0,c(4,n,2)-c(3,n,2): PLOT c(3,n+1,1)+d(n),c(4,
n+1,2): DRAW -d(n),0: DRAW 0,c(3,n+1,2)-c(4,n+1,2): D
RAW d(n),0
9090 NEXT n
9100 RETURN
9110 LET n=fal: LET fal=0: PLOT c(1,n,1),c(1,n,2): DR
AW 0,c(2,n,2)-c(1,n,2): DRAW c(3,n,1)-c(2,n,1),0: DRA
W 0,c(4,n,2)-c(3,n,2): DRAW c(1,n,1)-c(4,n,1),0: RETU
RN

```

Megjegyzés: Sajnos, akadnak programhibák a sok jó ötlet között. A 10-es nehézségi foknál, úgy látszik, végtelen ciklusba jut a program. Bizonyos esetekben könnyen „kikerül” a játékos a labirintusból, ilyenkor a látható ábra nem felel meg a labirintus térképének. A programlista nem tartalmaz megjegyzés- (REM-) sorokat, így a program kiegészítése, módosítása elég nehézkes, de lehet próbálkozni!

METEOR

Készítette: Dárday Zoltán

Leírás: Az a feladat, hogy a képernyőn felülről érkező csillagok között manőverezzünk úgy, hogy a csillagoknak nem szabad nekimenni. Az űrhajót a kurzorgombokkal lehet irányítani.

```
3 REM METEOR
5 CLEAR 29999
10 BORDER 1: PAPER 1: INK 6: CLS
30 PRINT AT 10,10;" * METEOR * "
60 RESTORE : DATA 16,16,56,56,108,124,108,40
70 FOR i=0 TO 7: READ a: POKE USR "#"+i,a: NEXT i
75 REM GEPI KOD BETOLTES
80 FOR i=0 TO 50: READ a: POKE 30000+i,a: NEXT i
90 LET Z=17: LET R=0
95 REM KEZDET
100 LET x=-24: INPUT "PRESS ""ENTER""", LINE q$
103 RANDOMIZE : CLS
105 REM FOLYTATAS
110 LET x=x+1
115 BEEP .008,10
130 IF SCREEN$ (23,z)<>"*" THEN GO TO 300
135 REM HIBA
140 BEEP .1,-5: BEEP .1,20
200 PRINT FLASH 1;AT 10,10;" SCORE ";x;" ";#0;AT 1,
z;" "
205 PRINT AT 5,10;" * METEOR * ": LET z=17
210 IF X>R THEN LET R=X: PRINT FLASH 1;AT 13,9;" N
EW"
220 PRINT FLASH 1;AT 13,13;" RECORD ";R;" "
230 GO TO 100
295 REM MOZGATAS
```

```


300 LET z=z+(INKEY$="8" AND z<31)-(INKEY$="5" AND z>
0)
310 PRINT #0;AT 1,z;" # "
320 LET a=INT (RND*32)
330 LET b=INT (RND*32)
340 LET c=INT (RND*32)
350 PRINT AT 0,a;"*";AT 0,b;"*";AT 0,c;"*"
360 LET a=USR 30000
400 GO TO 110
500 DATA 62,24,33,255,87,93,84,254,0,200,6,32,43,16,
-3,1,224,0,237,184,61,254,8,56,14,6,7,37,16,-3,1,32,0
,237,184,98,107,24,-32,6,32,8,62,0,18,27,16,-4,8,24,-
16
1000 SAVE "METEOR" LINE 1

```

Megjegyzés: A program bővíthető a csillagok számának növelésével, a haladási sebesség változtatásával, esetleg beépített objektumok elérése esetén adható jutalompontokkal. A pontozás több szinten is megoldható lenne, pl. ritka csillagképpel és lassan indulva az első szinten kevés pontot adva, később – növekvő pontszám mellett – növekedhetne a sebesség és az egy sorban elhelyezett csillagok száma. A színhatások és hangeffektusok is bővíthetők. Hiányosság, hogy „menet közben” nem jelzi a program az elért pontszámot (és a rekordot). A feliratok angol nyelvűek. Sajnos, használati utasítás még angolul sincs.

GUSZTÁV SPORTOL

Készítette: Dárday Zoltán



Leírás: A program három, egymás után következő játékot tartalmaz.

- **AKADÁLYFUTÓ:** Gusztávot a kurzorbillentyűkkel irányíthatjuk a felfelé haladó pálya nyílásain a „kapu” között. Három élet van;
- **KAPUS:** Gusztávnak 3X 10 labdát kell kivédenie;
- **HEGYMÁSZÓ:** a hegyvidék vonulatát a 6 (lefelé) és a 7 (felfelé) billentyűkkel kell nyomon követni: itt is három élet van.

```
1>REM GUSZTAV SPORTOL
2 RESTORE : FOR I=0 TO 7: READ B: POKE USR "A"+I,B
: NEXT I
3 LET R=0: DIM R$(5)
4 DIM J$(20,10): DIM J(20)
5 BORDER 4: PAPER 4: INK 7: CLS
6 GO TO 3000
7 LET O=0
8 CLS : PRINT FLASH 1;AT 11,4;" GUSZTAV AZ AKADAL
YFUTO ": PAUSE 300
9 FOR G=1 TO 3: CLS : LET X=4: LET A=-3
10 FOR I=1 TO 6: POKE 23692,255: PRINT AT 21,21,"■■■■
■■■■"
30 LET X=X+(INKEY$="B" AND X<6)-(INKEY$="5" AND X>1
)
```

```

50 PRINT INK 0; AT 0, X; " "
55 BEEP .01, 10
60 NEXT I
63 IF POINT (8*X+4, 165) THEN GO TO 100
65 LET A=A+1
70 PRINT AT 21, 21, "████████████████████"; AT 21, INT (RND*6)
+1; " "
80 BEEP .01, 10
85 GO TO 10
100 PRINT FLASH 1; AT 7, 11; " "; A; " KAPUN BUJTAL AT "
120 IF A>R THEN LET R=A: PRINT AT 13, 14; FLASH 1; "
UJ RECORD ": INPUT "IRD BE A NEVED!", LINE R$
130 PRINT AT 18, 10; "A RECORD: "; R; " "; R$
135 GO SUB 200
140 LET O=O+10*A
145 NEXT G
150 GO TO 1000
160 DATA 60, 60, 24, 255, 189, 189, 36, 102
200 BEEP .25, 0: BEEP .25, 2: BEEP .125, 3: BEEP .125, 2
: BEEP .25, 0
210 BEEP .285, 3: BEEP .25, 5: BEEP .25, 7
220 RETURN
1000 CLS : PRINT FLASH 1; AT 11, 7; " GUSZTAV A KAPUS "
: PAUSE 300
1001 FOR G=1 TO 3: CLS
1002 LET E=8: LET F=0: PRINT AT 13, 2; "████████████████████
████████████████████"; AT 13, E; INK 0; PAPER 7; " "
1020 FOR P=1 TO 10: LET W=INT (RND*13)+1: PRINT AT 16
, 0; P, F
1040 LET X=INT (RND*12)+3
1050 FOR I=1 TO 13: FOR K=1 TO W: NEXT K: PRINT AT I-
1, X; " "; AT I, X; "o"
1080 LET E=E+(INKEY$="8" AND E<14)-(INKEY$="5" AND E>
3)
1090 PRINT INK 0; PAPER 7; AT 13, E-1; " "
1100 NEXT I
1105 PRINT AT 13, X; "██████" AND X<>E
1110 IF X=E THEN LET F=F+1: BEEP .5, 10: GO TO 1120
1115 BEEP .5, -10

```

```

1120 NEXT P
1130 PRINT AT 16,0;"10",F: GO SUB 200
1150 LET D=D+10*F
1160 NEXT G
2000 CLS : PRINT FLASH 1;AT 11,4;" GUSZTAV A HEGYMAS
ZO ": PAUSE 300
2003 FOR G=1 TO 3: CLS
2010 LET Y=10: PRINT AT Y,0;"====="
2030 FOR X=5 TO 28: PRINT AT Y,X;"=";
2050 LET Y=Y+INT (RND*3)-1
2060 IF Y<5 THEN LET Y=5
2070 IF Y>15 THEN LET Y=15
2080 NEXT X: PRINT "CEL"
2100 LET Y=10: LET X=0
2104 PRINT AT Y;X;"="
2110 LET X=X+1: LET Y=Y+(INKEY$="6")-(INKEY$="7")
2140 LET N$=SCREEN$(Y,X)
2150 PRINT INK 0;AT Y,X;"■"
2160 BEEP .2,10
2170 IF N$="=" THEN GO TO 2104
2180 IF N$="C" THEN LET D=D+100
2200 LET D=D+10*X
2205 GO SUB 200: NEXT G: GO SUB 200
2210 CLS : PRINT AT 11,8;"PONTSZAMOD: ";D: PAUSE 100
2300 LET I=0
2310 LET I=I+1
2315 IF I=21 THEN GO TO 3000
2320 IF J(I)>0 THEN GO TO 2310
2325 IF I=20 THEN GO TO 2350
2330 FOR Z=20 TO I+1 STEP -1
2340 LET J(Z)=J(Z-1): LET J$(Z)=J$(Z-1): NEXT Z
2350 LET J(I)=0: LET J$(I)=" "
2360 CLS : PRINT TAB 3;"NEV","PONTSZAM",,,
2370 FOR W=1 TO 20: PRINT J$(W),
2371 IF J(W)<>0 THEN PRINT J(W)
2375 NEXT W
2380 PRINT AT I+1,0; FLASH 1;"■■■■"
2390 INPUT LINE J$(I)
2400 PRINT AT I+1,0;J$(I)

```

```


2405 PAUSE 150
3000 CLS : PRINT AT 3,24;"      ",TAB 2;"      "
      "      "
3001 PRINT TAB 2;"      "
      "      "
3002 PRINT TAB 2;"      "
      "      "
3003 PRINT TAB 2;"      "
      "      "
3004 PRINT TAB 2;"      "
      "      "
3005 PRINT TAB 2;"      "
      "      "
3006 PRINT AT 14,3;"      "
      "      "
3007 PRINT TAB 3;"      "
      "      "
3008 PRINT TAB 3;"      "
      "      "
3009 PRINT TAB 3;"      "
      "      "
3010 PRINT TAB 3;"      "
      "      "
3011 PRINT TAB 3;"      "
      "      "
3012 FOR I=0 TO 7: OUT 254,I: BEEP .02,10-I: IF INKEY
      $="" THEN NEXT I
3013 IF INKEY$="" THEN GO TO 3012
3014 GO TO 7
4000 SAVE "GUSZTAV SP" LINE 1

```

Megjegyzés: Jól megszerkesztett programsorozat. Itt különösen hiányzik a felhasználót tájékoztató útmutató felirat. Jó lenne, ha a három játék közül külön-külön bármelyik önállóan is indítható lenne. Ezenkívül (különösen a hegy-mászó programnál) játék közben is jelezni kellene az elért pontszámot.

GUSZTÁV MOTOROZIK

Készítette: Dárday Zoltán



Leírás: Úgy kell Gusztávval a kijelölt úton haladni, hogy az útszegélynek és az akadályoknak (bokor, kő, tábla) nem szabad nekimenni. Az irányítás a 6 (lefelé), ill. a 7 (felfelé) gombokkal történik. A kanna és a kapu felvételéért jutalompont jár.

```
2 REM GUSZTAV MOTOROZIK
3 CLEAR 29999: BEEP 1,0: LET R=0: BORDER 6: PAPER
4: INK 0: CLS
4 FOR i=0 TO 39: READ a: POKE USR "☐"+i,a: NEXT i
5 DATA 60,66,241,141,133,133,134,252
6 DATA 8,73,38,164,153,94,188,126
7 DATA 240,143,129,133,241,31,24,24
```



```

8 DATA 15,201,255,123,119,111,127,127
9 DATA 63,61,61,59,30,12,12,30
10 FOR I=0 TO 31: READ A: POKE USR "▲"+I,A: NEXT I
13 REM GEPI KOD BETOLTES
15 FOR i=30000 TO 30021: READ a: POKE i,a: NEXT i
20 DATA 0,3,3,7,14,31,63,60,0,128,128,128,0,0,240,5
6
30 DATA 124,127,123,101,164,164,136,112,16,240,222,
153,169,169,34,28
40 DATA 33,0,64,22,0,62,128,6,31,35,94,43,115,35,16
,249,114,35,61,32,242,201
41 REM KEZDET
42 GO SUB 700
43 BORDER 6: INK 0: CLS : PRINT AT 15,20,: FOR i=1
TO 6: PRINT PAPER 6,TAB 0: NEXT i
44 LET p=0: PRINT PAPER 6;AT 21,0;"SCORE: ";p,"HIG
H: ";R
45 FOR I=0 TO 20: PRINT AT 5,I;"=";AT 11,I;"=";: NE
XT I
48 LET Y=5: FOR I=21 TO 30
50 GO SUB 300
56 NEXT I
60 LET I=31: LET S=8
70 LET U=0: LET h=INT (RND*40)+50
95 REM FOLYTATAS
100 PRINT AT S,10;"▲ ";AT s+1,10;"●"
110 PRINT AT y+INT (RND*5)+1,31;"■" AND INT (RND*40
)=5
115 PRINT AT y+INT (RND*5)+1,31;"■" AND INT (RND*60
)=20
120 GO SUB 290
125 PRINT PAPER 6;AT 21,7;p
128 IF NOT POINT (101,170-s*8) AND NOT POINT (101,17
0-(s+1)*8) THEN GO TO 145
130 IF POINT (101,172-S*8) OR POINT (101,172-(S+1)*8
) THEN GO TO 400
135 IF POINT (102,173-s*8) OR POINT (102,173-(s+1)*8
) THEN GO SUB 500
140 IF POINT (98,175-s*8) OR POINT (98,175-(s+1)*8)

```



```

703 PRINT TAB 2;"          "
"          "
704 PRINT TAB 2;"          "
"          "
705 PRINT TAB 2;"          "
"          "
710 PRINT AT 16,6;"          "
"
711 PRINT TAB 6;"          "
712 PRINT TAB 6;"          "
725 FOR I=0 TO 7: OUT 254,I: BEEP .02,10-I: IF INKEY
$="" THEN NEXT I
730 IF INKEY$="" THEN GO TO 725
740 RETURN
1000 SAVE "GUSZTAV MO" LINE 1

```

UDG



Megjegyzés: Itt is jó lenne egy játékútmutató a képernyőre, bár a feladat elég közismert. „Menet közben” lehetne fokozni még a hanghatást. A pályák közötti „sivatagi” (?) terep ötletes. Valamilyen programhiba miatt a kupa nem mindig tűnik el, hiába szedi fel a játékos. Jó lenne, ha a „sivatagi” útszakasz után hibaként jelezné a program, hogy a motoros nem az útpadkák közé tért vissza.

AGYTORNA ÉS INTELLIGENCIATESZT

Készítette: Dárday Zoltán

Leírás 1: A program a közismert „Master mind” játék számítógépes változata. A gép öt egyjegyű számot rejt el, amelyek közül kétszer szerepelhet bármelyik számjegy (többször nem). Előre megadott számú próbálkozás során kell az öt számjegyet kitalálni. A program külön jelzi az eltalált helyű és értékű számjegyek darabszámát.

Leírás 2: Egy háromszög alakú rácson táblán egy hely üres (0). A többi 1-esekkel van feltöltve. Addig lehet lépni bármelyik 1-essel, amíg van átléphető szomszédos 1-es. Az a cél, hogy minél kevesebb 1-es maradjon a táblán. A lépéseket a koordináta-számpárokkal a „HONNAN-HOVÁ” kérdésre válaszolva lehet megadni. Ha már nincs több lépés, 0-val kell kilépni. Eztán derül ki az agytorna eredménye.

```
2 REM AGYTORNA
5 BORDER 1: PAPER 1: INK 6: CLS
6 READ B,C,D,E: DATA 200,100,50,30
8 PRINT FLASH 1;AT 11,11;" AGYTORNA ": PAUSE 0
10 PRINT FLASH 1;AT 11,6;" SUPER MASTER MIND "
20 PAUSE 250: CLS
30 PRINT 'TAB 5;"EGY OTJEGYU SZAMOT KELL",,TAB 2;"K
ITALALNOD, A SZAMJEGYEK 0-9",,TAB 7;"ERTEKUEK LEHETNE
K",AT 12,5;"** JO SZORAKOZAST! **"; FLASH 1;AT 17,5;"
NYOMJ MEG EGY GOMBOT "
40 PAUSE 0
50 CLS : INPUT TAB 5;"HANY LEHETOSEGET KERSZ?", LIN
E B#
52 LET H=50: GO SUB 5000
53 LET V=VAL B#
54 REM KOMBINACIO
55 RANDOMIZE : FOR N=1 TO 5
58 LET Z=INT (RND*10)
```

```

60 IF N=1 THEN LET A=Z
70 IF N=2 THEN LET B=Z
72 IF B=A THEN GO TO 58
80 IF N=3 THEN LET C=Z
82 IF C=A OR C=B THEN GO TO 58
90 IF N=4 THEN LET D=Z
92 IF D=A OR D=B OR D=C THEN GO TO 58
100 IF N=5 THEN LET E=Z
102 IF E=A OR E=B OR E=C OR E=D THEN GO TO 58
110 NEXT N
130 CLS
140 PRINT AT 0,4;"** SUPER MASTER MIND **",,,V;TAB 6
; "JO SZAM";TAB 22;"JO POZICIO",,
150 IF V<1 THEN GO TO 1300
160 LET X=-1: LET Y=-1
180 INPUT "TIPP?", LINE B$
182 IF LEN B$<>5 THEN GO TO 180
190 LET H=180: GO SUB 5000
200 LET A1=VAL B$(1): LET B1=VAL B$(2): LET C1=VAL B
$(3): LET D1=VAL B$(4): LET E1=VAL B$(5)
250 PRINT AT 21,14;B$
255 IF A1=A THEN GO SUB 1000
260 IF B1=B THEN GO SUB 1000
265 IF C1=C THEN GO SUB 1000
270 IF D1=D THEN GO SUB 1000
275 IF E1=E THEN GO SUB 1000
280 IF A<>A1 AND (A=B1 OR A=C1 OR A=D1 OR A=E1) THEN
GO SUB 1100
285 IF B<>B1 AND (B=A1 OR B=C1 OR B=D1 OR B=E1) THEN
GO SUB 1100
290 IF C<>C1 AND (C=A1 OR C=B1 OR C=D1 OR C=E1) THEN
GO SUB 1100
295 IF D<>D1 AND (D=A1 OR D=B1 OR D=C1 OR D=E1) THEN
GO SUB 1100
300 IF E<>E1 AND (E=A1 OR E=B1 OR E=C1 OR E=D1) THEN
GO SUB 1100
310 LET V=V-1
315 PRINT AT 21,28
320 POKE 23692,19: PRINT

```

```

325 BEEP .5,0
330 GO TO 140
1000 LET Y=Y+1: PRINT AT 21,22+Y;"*"
1020 IF Y=4 THEN CLS : GO TO 1500
1030 RETURN
1100 LET X=X+1: PRINT AT 21,6+X;"+": RETURN
1300 CLS
1310 PRINT AT 11,5;"** EZ NEM SIKERULT! **"
1320 PRINT AT 15,7;"A MEGOLDAS: ";A;B;C;D;E
1430 PAUSE 0: RUN 3000
1500 PRINT FLASH 1;AT 11,10;" GRATULALOK "
1630 GO TO 1320
3000 PRINT FLASH 1;AT 11,9;" MEMORIATESZT ": PAUSE 2
50: CLS
3001 PRINT "ELOSZOR 1 MAJD 2,3,4,..20 SZAMOT",TAB 3;"
FOGSZ LATNI. EZEKET KELL",,TAB 5;"A LATOTTAKKAL MEG
EGYEZO ",,TAB 4;"SORRENDEN A GEPBE BEIRNI",AT 12,5;"
** JO SZORAKOZAST! **"; FLASH 1;AT 17,5;" NYOMJ MEG E
GY GOMBOT "
3009 RANDOMIZE : PAUSE 0: CLS : PAUSE 80
3010 LET A#=CHR# INT (RND*10+48): LET P=0
3020 LET P=P+1
3040 FOR X=1 TO P: PRINT AT 0,CODE A#(X)-48;A#(X): PA
USE 50: CLS : NEXT X
3090 INPUT LINE B#: BEEP .5,0
3100 IF A#<>B# THEN GO TO 3160
3110 LET A#=A#+CHR# INT (RND*10+48)
3120 IF P<20 THEN GO TO 3020
3130 PRINT AT 11,8;"** GRATULALOK **"
3140 PAUSE 0: RUN 4000
3160 PRINT AT 11,8;"FIGYELJ JOBBAN!";AT 16,6;A#;AT 19
,6;B#
3170 GO TO 3140
3800 PRINT FLASH 1;AT 20,7;"HIBA": PAUSE 100: PRINT
AT 20,7;" ": RETURN
4000 PRINT FLASH 1;AT 11,5;" INTELLIGENCIA TESZT ":
PAUSE 250: CLS
4001 PRINT "A JATEK LENYEGE, HOGY A GULABAN",,TAB 3;"

```

```

A LEHETO LEGKEVESEBB 1-ES",,TAB 3;"MARADJON. ATLEPETT
  1-ESBOL",,TAB 3;"0 LESZ, CSAK 0-RA LEPHETSZ";AT 11,6
;"1 1 0    =>    0 0 1";AT 16,5;"** JO SZORAKOZAST! *
*";AT 20,5; FLASH 1;" NYOMJ MEG EGY GOMBOT "
4009 PAUSE 0
4010 CLS : LET P=14: LET K=5
4060 PRINT AT 2,K;"0";AT 10,K+17;"11"
4070 FOR I=2 TO 5: LET K=9-K
4100 FOR J=1 TO I: LET H=10*I+J
4120 PRINT AT 2*I,K;"1";AT 2*I+8,2*K+12;H
4130 LET K=K+2: NEXT J
4150 LET K=K-2: NEXT I
4190 INPUT "HONNAN?", LINE B$
4191 LET H=4190: GO SUB 5000
4193 LET B=VAL B$
4195 IF B=0 THEN GO TO 4800
4200 LET I=INT (B/10): LET J=B-10*I
4220 IF J>I OR I>5 OR I<1 OR J<1 THEN GO SUB 3800: G
O TO 4190
4225 IF SCREEN$ (2*I,2*J-I+4)<>"1" THEN GO SUB 3800:
GO TO 4190
4228 INPUT "HOVA?", LINE B$: LET H=4228: GO SUB 5000
4229 LET C=VAL B$
4230 LET X=INT (C/10): LET Y=C-10*X
4250 IF Y>X OR X>5 OR Y<1 OR X<1 THEN GO SUB 3800: G
O TO 4228
4260 IF SCREEN$ (2*X,2*Y-X+4)<>"0" THEN GO SUB 3800:
GO TO 4228
4270 IF SCREEN$ (2*I,2*J-I+6)<>"0" AND C=B+2 THEN GO
TO 4700
4280 IF SCREEN$ (2*I,2*J-I+2)<>"0" AND C=B-2 THEN GO
TO 4700
4290 IF SCREEN$ (2*I+2,2*J-I+3)<>"0" AND C=B+20 THEN
GO TO 4700
4300 IF SCREEN$ (2*X+2,2*Y-X+3)<>"0" AND C=B-20 THEN
GO TO 4700
4310 IF SCREEN$ (2*I+2,2*J-I+5)<>"0" AND C=B+22 THEN
GO TO 4700

```

```

4320 IF SCREEN$(2*X+2,2*Y-X+5)<>"0" AND C=B-22 THEN
  GO TO 4700
4330 GO SUB 3800
4340 GO TO 4190
4700 PRINT AT 2*I,2*J-I+4;"0";AT 2*X,2*Y-X+4;"1"
4705 BEEP .5,0
4710 LET Q=B+(C-B)/2: LET I=INT (Q/10): LET J=Q-10*I
4720 PRINT AT 2*I,2*J-I+4;"0"
4730 LET P=P-1
4750 IF P=1 THEN CLS : PRINT AT 11,11;"ZSENIALIS": P
  AUSE 0: RUN
4760 GO TO 4190
4800 CLS
4810 PRINT AT 11,11;"NAGYON JO" AND P=2
4820 PRINT AT 11,11;"MEGFELELO" AND P=3
4830 PRINT AT 11,13;"GYENGE" AND P=4
4850 PRINT AT 11,10;"NAGYON GYENGE" AND P>4
4900 PAUSE 0: RUN
5000 IF CODE B$=0 THEN GO TO H
5010 FOR E=1 TO LEN B$: IF B$(E)<"0" OR B$(E)>"9" THE
  N GO TO H
5020 NEXT E
5050 RETURN

```

Megjegyzés 1: A program, bár fejtörő jellegű, jobban is kihasználhatná a gép hang- és főként színkezelését. Úgy tűnik, az ismételt számjegyek esetén a kijelzés algoritmusá eltér a szokásos játék szabályaitól. Ezt vagy külön fel kellene tüntetni, vagy mint programhibát – ki kell javítani. Elég kevés a REM az amúgy is zsúfolt listában.

Megjegyzés 2: A félrevezető megnevezéstől eltekintve (pl. solitary-ként lehet ismerős) jól megírt és szépen kivitelezett program.–Sajnos túlzásúfolt és megjegyzésekkel alig ellátott a programlista. A szín- és hangkezelés is lehetne ötletesebb.

VADÁSZAT

Készítette: Mészáros Péter

Pontok : 48



REKORD : 0



Leírás: A képernyő felső részén egy madár mozog, amit el kell találni. A vadász minden találatáért 12 pont jár. Vigyázni kell azonban, mert a madár is tud lőni. A játékos a képernyő alsó szélén mozogva manőverezhet.

```
1 REM MESZAROS PETER 1984<
2 BORDER 2: PAPER 2: INK 7: CLS
3 GO SUB 1000
5 LET r=0
8 GO SUB 6000
10 LET x=INT (RND*29)+1
```

```

15 LET w=10
20 LET y=INT (RND*31)
25 LET p=0
30 LET h=0
40 LET j=0
50 PRINT AT 2,0;"          "
"          "
60 PRINT AT 0,0;"Pontok:";p
70 PRINT AT 0,20;"REKORD:";r
80 FOR u=1 TO 3-j: PRINT AT 0,14+u;"  " : NEXT u
100 IF INKEY$="6" AND x>0 THEN LET x=x-1
105 IF INKEY$="0" THEN GO SUB 400
110 IF INKEY$="7" AND x<28 THEN LET x=x+1
115 GO SUB 5000
120 PRINT AT 21,x;"  "
130 LET f=INT (RND*w)
135 IF w>2 THEN LET w=w-.01
140 IF f=1 THEN GO SUB 2000
200 GO TO 100
400 FOR l=1 TO 2: OVER 1: PLOT x*8+16,6: DRAW 0,137
: OVER 0: NEXT l
410 BEEP .01,0
420 IF x=y OR x+1=y AND INKEY$="0" THEN GO TO 500
430 RETURN
500 FOR k=0 TO 10
510 BEEP .01,y+k
520 PRINT AT 3,y+1;"  "
525 IF y<20 THEN LET y=y+5: GO TO 530
527 IF y>5 THEN LET y=y-5
530 NEXT k
540 LET p=p+12
545 IF p=700 THEN LET j=j-1
550 CLS : GO TO 50
1000 RESTORE
1005 FOR i=USR "a" TO USR "h"-1
1010 READ s
1020 POKE i,s
1030 NEXT i
1035 GO TO 4000

```

```

1040 DATA 1,1,BIN 00000011,BIN 00100111,BIN 00100111,
BIN 01111111,BIN 01111100,BIN 11111001
1050 DATA 128,128,BIN 11000000,BIN 11100100,BIN 11100
100,BIN 11111110,BIN 00111110,BIN 10011111
1060 DATA 0,0,BIN 01000001,BIN 01100011,127,BIN 00111
110,BIN 00011100,BIN 01101011
1070 DATA 195,195,0,0,24,0,195,195
1080 DATA 16,16,16,56,BIN 11111110,BIN 11111110,198,1
46
1090 DATA 16,16,124,68,124,68,68,0
1100 DATA 16,16,124,64,112,64,124,0
2000 FOR 1=1 TO 2: OVER 1: PLOT y*8+16,138: DRAW 0,-1
37: OVER 0: NEXT 1
2010 IF y=x OR y=x+1 AND f=5 THEN GO TO 2030
2020 RETURN
2030 PRINT FLASH 1;AT 10,12;"bumm!!!!"
2032 PRINT AT 21,x+1;"  "
2035 PRINT AT 21,x+1;"  "
2038 PRINT AT 21,x+1;"  "
2040 BEEP .5,-10
2045 FOR 1=1 TO 2: OVER 1: PLOT y*8+16,138: DRAW 0,-1
37: OVER 0: NEXT 1
2050 LET j=j+1
2060 IF j<3 THEN CLS : GO TO 50
2070 PRINT AT 17,5;"meghaltal!"
2075 BEEP .05,30: BEEP .05,10
2100 IF p<=36 THEN PRINT FLASH 1;AT 19,3;"Kezdo!..K
ezdj magaddal mast!"
2110 IF p>36 AND p<=100 THEN PRINT FLASH 1;AT 19,0;
"Javulsz,de csereltesd a reflexe-idet!"
2120 IF p>100 AND p<=300 THEN PRINT FLASH 1;AT 19,0
;"Van benned tehetség,de ehhez keves!"
2130 IF p>300 AND p<=1000 THEN PRINT FLASH 1; BRIGH
T 1;AT 19,0;"Egesz jól csinalod!"
2135 PRINT AT 0,13;" "
2140 IF p>1000 AND p<=1500 THEN PRINT FLASH 1;AT 19
,0;"Sztar vagy!"
2150 IF p>1500 THEN PRINT FLASH 1;AT 19,0;"Igazi me
stervadasz vagy! ";p/12;" madarat kilott
el!": GO SUB 3000

```


```

2180 IF p>r THEN PRINT AT 8,0;"Ez a rekord!": LET r=
p
2200 PRINT AT 3,0;"UJ JATEK?(i/n)"
2210 IF INKEY$="i" OR INKEY$="I" THEN CLS : GO TO 10
2220 IF INKEY$="n" OR INKEY$="N" THEN STOP
2230 GO TO 2110
3000 BEEP .5,25: BEEP .2,15: BEEP .3,10: BEEP .4,20:
BEEP .2,15: BEEP .3,10: BEEP .7,25
3010 RETURN
4000 CLS : PRINT AT 0,8;"V a d a s z a t"
4010 PRINT AT 4,0;" ~~~~~~
~~~~~"
4020 PRINT AT 18,0;" ~~~~~~
~~~~~"
4030 PRINT AT 2,8;"JATEKSZABALYOK"
4040 PRINT AT 7,0;" A jatekter felső részen mo
zog egy madár, amit el ~~~~~~
~~~~~ kell találnod. Erre 12
pontot kapsz. DE jól vi- ~~~~~~
~~~~~ gyazz! Ő is
éppúgy elta- ~~~~~~
~~~~~ lálhat! Fegyvered az also ~~~~~~
~~~~~
terfelen mozgatható."
4050 PRINT AT 16,10;"JÓ JATEKOT!"
4060 PRINT AT 20,5;"Nyomj meg egy gombot!"
4070 PAUSE 4E4
4080 CLS
4100 PRINT AT 4,0;" ~~~~~~
~~~~~"
4110 PRINT AT 18,0;" ~~~~~~
~~~~~"
4120 PRINT AT 2,5;"IRANYÍTÁS:"
4130 PRINT AT 15,0;" <-- 6 ----- 7 --> "
4140 PRINT AT 10,13;"loves"
4150 PRINT AT 8,15;"0"
4160 PRINT AT 6,15;"^"
4170 PRINT AT 20,5;"Nyomj meg egy gombot!"
4180 PAUSE 4E4
4190 RETURN
5000 REM
5050 BEEP .001,20
5060 LET z=INT (RND*2)

```


CHALLENGER

Készítette: Mészáros Péter



Leírás: Egy UFO-t kell irányítani a rakéták között az űrbázishoz, lehetőleg ütközések nélkül. 200 pontonként két új UFO-t ad a program. A 0-val előre, a 6-tal visszafelé lehet az űrhajót mozgatni. A játék tetszőleges billentyű megnyomásakor elindul.

```
1 REM MÉS ZÁR OS PÉ TER 1984<
2 CLS : LET h=0
3 GO SUB 9000
4 DIM q(6,2): GO SUB 8500
5 GO SUB 4000
6 REM ORA JATEK
7 LET p=0
8 LET f=0
```

```

9 LET f=0
10 LET u=0
11 LET g=0
12 PAPER 3: BORDER 3: INK 7: CLS
13 LET r=4: PRINT BRIGHT 1;AT 5,17;"-----"
14 LET t=1
20 GO SUB 8000
25 LET k=0
30 LET s=5
35 IF (p/200)-INT (p/200)>u AND p<>0 THEN LET r=r+
(4-r): LET u=u+1: FOR d=1 TO r: PRINT BRIGHT 1;AT 5,
16+r;"-": NEXT d
37 LET a=INT (RND*4)+1
38 LET b=INT (RND*4)+1
39 IF a=b THEN GO TO 38
40 FOR v=0 TO 10 STEP 2
45 PRINT AT 19,20; BRIGHT 1;p
50 GO SUB 2000
70 PRINT AT q(t,1),q(t,2);".."
80 IF INKEY$="0" AND t<6 THEN PRINT AT q(t,1),q(t,
2);" ": LET g=0: LET t=t+1: BEEP .01,40: IF CODE SCRE
EN$ (q(t,1),q(t,2))<>32 THEN GO SUB 3000
90 IF INKEY$="6" AND t>1 THEN PRINT AT q(t,1),q(t,
2);" ": LET g=0: LET t=t-1: BEEP .01,42: IF CODE SCRE
EN$ (q(t,1),q(t,2))<>32 THEN GO SUB 3000
100 IF INKEY$<>"6" AND INKEY$<>"0" AND t=1 THEN LET
g=g+1
110 IF g>=15 THEN LET t=t+1: BEEP .01,30: LET g=0:
PRINT AT 10,2;" "
150 IF a=1 OR b=1 AND v=0 THEN PRINT AT 5,4;"#"
160 IF a=2 OR b=2 AND v=0 THEN PRINT AT 16,7;"#"
170 IF a=3 OR b=3 AND v=0 THEN PRINT AT 5,11;"#"
180 IF a=4 OR b=4 AND v=0 THEN PRINT AT 16,14;"#"
190 IF v<>0 THEN GO SUB 1000
200 IF a=1 OR b=1 THEN PRINT AT 5+v,4;"#"
210 IF a=2 OR b=2 THEN PRINT AT 16-v,7;"#"
220 IF a=3 OR b=3 THEN PRINT AT v+5,11;"#"
230 PRINT AT 16-v,14;"#"

```



```

240 IF v=0 THEN PRINT AT 15,4;" ";AT 6,7;" ";AT 15,
11;" ";AT 6,14;" "
250 IF t=6 THEN GO SUB 2000
270 IF INKEY$="6" AND t>1 THEN PRINT AT q(t,1),q(t,
2);" ": LET g=0: LET t=t-1: BEEP .01,52: IF CODE SCRE
EN$ (q(t,1),q(t,2))<>32 THEN GO SUB 3000
700 NEXT v
710 IF t=a+1 OR t=b+1 THEN BEEP .8,0: PRINT AT q(t,
1),q(t,2);"#": GO SUB 3000
720 IF f=18 THEN LET f=0
730 IF f=15 THEN LET k=0
750 IF f<18 THEN LET f=f+1
998 BEEP .005,60
999 GO TO 30
1000 REM RAKETAK
1010 PRINT AT 3+v,4;" "
1020 PRINT AT 3+v,11;" "
1030 PRINT AT 18-v,7;" "
1040 PRINT AT 18-v,14;" "
1499 RETURN
2000 REM B#ZIS
2005 IF k=0 THEN PRINT AT 10,18;" ";AT 11,18;" "
2010 IF k=1 THEN PRINT AT 10,18;" .";AT 11,18;" "
2020 IF k=2 THEN PRINT AT 11,18;" " ;AT 10,18;" .
"
2023 IF f>14 THEN PRINT BRIGHT 1; FLASH 1;AT 10,18;
" ";AT 11,18;" " : LET k=0
2025 IF t<>6 THEN RETURN
2030 LET p=p+2
2033 IF f>14 AND t=6 THEN LET k=k-1
2035 LET t=1
2040 IF k<2 THEN LET k=k+1: RETURN
2050 PRINT AT 10,18;" . . .";AT 11,18;" "
2055 IF f=15 THEN PRINT BRIGHT 1; FLASH 1;AT 10,18;
" ";AT 11,18;" " : LET k=0
2059 FOR o=1 TO 4: BEEP .6,50: PAUSE 10: NEXT o
2060 LET k=0
2070 LET p=p+10
2075 PRINT BRIGHT 1;AT 19,20;p

```

```

2080 RETURN
3000 REM UTKOZES
3010 BEEP .05,20: BEEP .05,20: BEEP .05,20: BEEP .1,2
5: PAUSE 10: BEEP .05,20: BEEP .05,20: BEEP .05,20: B
EEP .05,15
3020 PRINT AT q(t,1),q(t,2);"30"
3030 LET r=r-1
3035 LET t=1
3040 PRINT AT 5,17;"    "
3050 FOR d=1 TO r: PRINT BRIGHT 1;AT 5,16+d;"-": NE
XT d
3060 IF r>0 THEN RETURN
3070 PRINT AT 20,23;"vege"
3080 PRINT BRIGHT 1;AT 5,20;"Potszam:";p
3090 IF p>h THEN PRINT BRIGHT 1;AT 7,20;"Ez a rekor
d": BEEP .01,30: BEEP .02,20: BEEP .01,25: LET h=p
3095 PRINT AT 0,26;"R:";h
3100 PRINT FLASH 1; BRIGHT 1;AT 10,20;"Uj jatek?"
3110 IF INKEY$="i" OR INKEY$="I" THEN GO TO 6
3120 IF INKEY$="n" OR INKEY$="N" THEN STOP
3130 GO TO 3110
4000 REM JATEK SZABALY
4010 GO SUB 8000
4020 FOR i=1 TO 6: PRINT AT q(i,1),q(i,2);"=": BEEP
.001,40: NEXT i
4030 PRINT AT 11,18;"szabaly"
4040 PRINT AT 0,21;"Z I N G O"
4050 PRINT AT 2,21;" Egy ufot =";AT 3,21;"kell irány
i";AT 4,21;"tanod a ra";AT 5,21;"ketak kozta";AT 6,2
1;"lehetőleg";AT 7,21;"utkozések";AT 8,21;"nelkül,hog
y";AT 9,21;"élj a ba";AT 10,21;"zishoz. szabaly";AT 11
,21;"200 ponton";AT 12,21;"kett kapsz";AT 13,21;"ket
ufot.Jo";AT 14,21;"jatekot! '0'";AT 15,21;"-val előre"
;AT 16,21;"'6-tal vissz";AT 17,21;"szahuzhatod";AT 18,
21;"START=barmi";AT 19,21;"ilyen gomb."
4060 PAUSE 4E4
4070 RETURN
8000 REM PALYA
8010 PLOT 24,103: DRAW 0,33: DRAW 33,0

```

```

8020 PLOT 96,135: DRAW 33,0: DRAW 0,-33
8030 PLOT 24,72: DRAW 0,-33: DRAW 33,0
8040 PLOT 96,39: DRAW 33,0: DRAW 0,33
8050 PLOT 48,96: DRAW 0,17: DRAW 17,0
8060 PLOT 88,111: DRAW 17,0: DRAW 0,-17
8070 PLOT 48,81: DRAW 0,-17: DRAW 17,0
8080 PLOT 88,63: DRAW 17,0: DRAW 0,17
8090 PLOT 16,88: DRAW 32,0: DRAW 0,-1: DRAW 28,0: DRA
W 0,1: DRAW 28,0: DRAW 0,-1: DRAW 22,0
8100 PLOT 76,24: DRAW 0,128
8110 RETURN
8500 REM TANK
8510 LET q(1,1)=10: LET q(1,2)=2
8520 LET q(2,1)=15: LET q(2,2)=4
8530 LET q(3,1)=6: LET q(3,2)=7
8540 LET q(4,1)=15: LET q(4,2)=11
8550 LET q(5,1)=6: LET q(5,2)=14
8560 LET q(6,1)=11: LET q(6,2)=16
8590 RETURN
9000 REM UDG
9010 DATA 56,214,124,124,56,56,56,16
9020 DATA 16,56,56,56,124,124,214,56
9030 DATA 0,0,124,0,124,254,84,0
9040 DATA 7,10,255,223,255,0,15,7
9050 DATA 192,160,254,246,254,0,224,192
9060 DATA 231,10,255,223,255,0,15,7
9070 DATA 206,160,254,246,254,0,224,192
9080 DATA 0,0,0,0,1,0,67,0
9090 DATA 0,0,0,0,0,0,132,0
9100 DATA 0,0,0,0,0,0,64,0
9110 DATA 0,0,0,0,1,0,3,0
9120 DATA 0,0,0,0,0,0,128,0
9130 DATA 0,0,0,0,0,0,4,0
9140 DATA 195,129,36,0,36,0,129,195
9150 DATA 0,0,0,126,126,0,0,0
9160 DATA 16,16,124,68,124,68,68,0
9170 DATA 16,16,124,64,112,64,124,0
9500 FOR i=USR "a" TO USR "r"-1
9510 READ s

```


A tanulságokat összegyűjtve, próbáljuk meg összefoglalni azokat a tanácsokat, amelyek leginkább segíthetik az olvasó-programozókat!

1. A program bőven el legyen látva REM-ekkel!
 2. Az utasítások sorszámozása lehetőleg 10-esével történjék!
 3. Egy sorba csak az összetartozó utasítások kerüljenek!
 4. Ismétlődő és/vagy logikailag elkülöníthető programrészek szubrutinba valók!
 5. Legyenek olyan változónevek, amelyek utalnak a logikai funkcióra!
 6. Az összetett aritmetikai kifejezések legyenek zárójelezve!
 7. Lehetőleg készüljön a program mellé folyamatábra is!
 8. Képernyő-kiírások tájékoztassák a program felhasználóját a legfontosabb tudnivalókról!
 9. A kiírások legyenek esztétikusak, szellemesek, és nehogy legyen bennük helyesírási vagy fogalmazási hiba (inkább többször nézze át más is)!
 10. Érdeemes megfogadni (de legalábbis megfontolni) egy-egy mérvadónak ítélt személy jótanácsait: egy kívülálló mindig tisztábban látja a problémákat!
- Ezt a „Tízparancsolat”-ot persze nem szabad kizárólagosnak tekinteni. Egy mikroszámítógép esetén sokszor kell jó érzékkel megkeresni a helyes közép-utat a takarékos helykihasználás, a program sebessége és egyéb szolgáltatásai között. Legtöbbször nem is lehet általános szabályt vagy tanácsot mondani, a szükségszerűség — többnyire a korlátozott memória — gyakran parancsoló kényszerhelyzeteket teremt. Mégis úgy érezzük, hogy a még kezdő programozók számára ez a néhány jó tanács lehetőséget ad egy „profibb” munkastílus kialakítására. Ráadásul ezek — legalábbis szellemükben, szándékukban — nemcsak a BASIC programok írása esetén érvényesek, hanem más programnyelvek, de más számítógépek esetén is javítják az elkészült programok minőségét.
- Még egyszer, jó szórakozást és eredményes munkát kívánunk minden programozónak a játékok felhasználásához, továbbfejlesztéséhez és használatához. Végül egy felhívás: várjuk azok jelentkezését, akik kedvet éreznek önálló játékprogramok írására, és munkájukat szívesen közkinccsé teszik. Elkövetkező kiadványainkban folytatni szeretnénk hasonló listák közlését a Magyarországon legelterjedtebb további számítógépekre (HT-1080Z, Commodore stb.). A jelentkezéseket a Mikro Magazin szerkesztőségébe várjuk.

Kiadja a Műszaki Könyvkiadó
Felelős kiadó: Fischer Herbert igazgató
A szedés a Műszaki Könyvkiadó Composer részlegében készült
Révai, Eger F. v.: Horváth Józsefné dr.

Műszaki vezető: Kőrizs Károly
Fedélterv és tipográfia: Lovász György
A könyv formátuma: A 5
Ívterjedelme: 5 ív (A5)
Ábrák száma: 12
Papír minősége: 100 g ofszet
Betűcsalád és -méret: Univers 10/11
Azonossági szám: 61 457
MÚ: 3600-h-8588
Készült az MSZ 5601 és 5602 szerint
Kézirat lezárva 1985. július